# Calculations with finite groupoids and their homomorphisms

1.71

7 August 2022

Emma J. Moore

**Chris Wensley** 

#### **Chris Wensley**

Email: c.d.wensley@bangor.ac.uk

Homepage: https://github.com/cdwensley

Address: Dr. C.D. Wensley

School of Computer Science and Electronic Engineering

Bangor University Dean Street Bangor

Gwynedd LL57 1UT

UK

## **Abstract**

The groupoids package provides functions for computation with groupoids (categories with every arrow invertible) and their morphisms; for graphs of groups, and graphs of groupoids. The most basic structure introduced is that of *magma with objects*, followed by *semigroup with objects*, then *monoid with objects* and finally *groupoid* which is a *group with objects*.

It provides normal forms for Free Products with Amalgamation and for HNN-extensions when the initial groups have rewrite systems and the subgroups have finite index. This is described in Section 6.2. It is planned to move this section to a new package Rewriting in time for version 4.11 of GAP.

The groupoids package was originally implemented in 2000 (as GraphGpd) when the first author was studying for a Ph.D. in Bangor.

The package was then renamed Gpd and version 1.07 was released in July 2011, ready for GAP 4.5.

Gpd became an accepted GAP package in May 2015.

In April 2017 the package was renamed again, as groupoids.

Recent versions implement many of the constructions described in the paper [AW10] for automorphisms of groupoids.

Bug reports, comments, suggestions for additional features, and offers to implement some of these, will all be very welcome.

Please submit any issues at https://github.com/gap-packages/groupoids/issues/ or send an email to the second author at c.d.wensley@bangor.ac.uk.

## Copyright

© 2000-2019, Emma Moore and Chris Wensley.

The groupoids package is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version.

## Acknowledgements

This documentation was prepared using the GAPDoc [LN17] and AutoDoc [GH17] packages.

The procedure used to produce new releases uses the package GitHubPagesForGAP [Hor17] and the package ReleaseTools.

# **Contents**

1 Introduction							
2	Many-object structures						
	2.1	Magmas with objects; arrows	<b>6</b>				
	2.2	Semigroups with objects	9				
	2.3	Monoids with objects	10				
	2.4	Generators of magmas with objects	10				
	2.5	Structures with more than one piece	11				
3	Map	Mappings of many-object structures 13					
	3.1	Homomorphisms of magmas with objects	13				
	3.2	Homomorphisms of semigroups and monoids with objects	15				
	3.3	Homomorphisms to more than one piece	16				
	3.4	Mappings defined by a function	17				
4	Groupoids 18						
	4.1	Groupoids: their properties and attributes	18				
	4.2	Groupoid elements; stars; costars; homsets	24				
	4.3	Subgroupoids	26				
	4.4	Left, right and double cosets	30				
	4.5	Conjugation	32				
	4.6	Groupoids formed using isomorphisms	33				
	4.7	Groupoids whose objects form a monoid	34				
5	Homomorphisms of Groupoids 37						
	5.1	Homomorphisms from a connected groupoid	37				
	5.2	Properties and attributes of groupoid homomorphisms	39				
	5.3	Special types of groupoid homomorphism	40				
	5.4	Homomorphisms to a connected groupoid	43				
	5.5	Homomorphisms to more than one piece	45				
	5.6	Automorphisms of groupoids	46				
	5.7	Matrix representations of groupoids	53				
6	Graphs of Groups and Groupoids 55						
	6.1	Digraphs	55				
	6.2	Graphs of Groups	56				
	6.3	Words in a Graph of Groups and their normal forms	58				

groupoids		4

	6.4	Free products with amalgamation and HNN extensions	59			
	6.5	GraphsOfGroupoids and their Words	62			
7	Technical Notes					
	7.1	Many object structures	65			
	7.2	Many object homomorphisms	67			
8	Development History					
	8.1	Versions of the Package	68			
	8.2	What needs to be done next?	69			
Re	References					
In	Index					

# **Chapter 1**

## Introduction

Groupoids are mathematical categories in which every arrow is invertible. The groupoids package provides functions for the computation with groupoids and their morphisms; for graphs of groups and graphs of groupoids. The package is far from complete, and development continues.

It was used by Emma Moore in her thesis [Moo01] to calculate normal forms for *free products* with amalgamation, and for HNN-extensions when the initial groups have rewriting systems.

The package may be obtained as a compressed tar file groupoids-version.number.tar.gz by ftp from one of the following sites:

- the groupoids GitHub site: https://github.com/gap-packages.github.io/groupoids/.
- any GAP archive, e.g. https://www.gap-system.org/Packages/packages.html;

The package also has a GitHub repository at: https://github.com/gap-packages/groupoids/.

The information parameter InfoGroupoids takes default value 1 which, for the benefit of new users, causes more messages to be printed out when operations fail. When raised to a higher value, additional information is printed out.

Help is available in the usual way.

```
gap> LoadPackage( "groupoids" );
```

For version 1.05 the package was completely restructured, starting with *magmas with objects* and their mappings, and building up to groupoids via semigroups with objects and monoids with objects. From version 1.07 the package includes some functions to implement constructions contained in [AW10]. More functions will be released as soon as possible.

Once the package is loaded, it is possible to check the correct installation by running the test suite of the package with the command ReadPackage("groupoids","tst/testing.g"); . Additional tests may be run using ReadPackage("groupoids","tst/testextra.g"); . (The file "tst/testall.g" is used for automated testing.)

You may reference this package by mentioning [BMPW02], [Moo01] and [AW10].

Additional information can be found on the *Computational Higher Dimensional Algebra* website at: https://github.com/cdwensley.

# **Chapter 2**

# Many-object structures

The aim of this package is to provide operations for finite groupoids. A *groupoid* is constructed from a group and a set of objects. In order to provide a sequence of categories, with increasing structure, mimicing those for groups, we introduce in this chapter the notions of *magma with objects*; *semigroup with objects* and *monoid with objects*. The next chapter introduces morphisms of these structures. At a first reading of this manual, the user is advised to skip quickly through these first two chapters, and then move on to groupoids in Chapter 3.

For the definitions of the standard properties of groupoids we refer to P. Higgins' book "Categories and Groupoids" [Hig05] (originally published in 1971, reprinted by TAC in 2005), and to R. Brown's book "Topology" [Bro88], recently revised and reissued as "Topology and Groupoids" [Bro06].

## 2.1 Magmas with objects; arrows

A magma with objects M consists of a set of objects Ob(M), and a set of arrows Arr(M) together with tail and head maps  $t,h: Arr(M) \to Ob(M)$ , and a partial multiplication  $*: Arr(M) \to Arr(M)$ , with a\*b defined precisely when the head of a coincides with the tail of b. We write an arrow a with tail u and head v as  $(a: u \to v)$ .

When this multiplication is associative we obtain a *semigroup with objects*.

A *loop* is an arrow whose tail and head are the same object. An *identity arrow* at object u is a loop  $(1_u : u \to u)$  such that  $a * 1_u = a$  and  $1_u * b = b$  whenever u is the head of a and the tail of b. When M is a semigroup with objects and every object has an identity arrow, we obtain a *monoid with objects*, which is just the usual notion of mathematical category.

An arrow  $(a: u \to v)$  in a monoid with objects has *inverse*  $(a^{-1}: v \to u)$  provided  $a*a^{-1} = 1_u$  and  $a^{-1}*a = 1_v$ . A monoid with objects in which every arrow has an inverse is a *group with objects*, usually called a *groupoid*.

#### 2.1.1 MagmaWithObjects

▷ MagmaWithObjects(args) (function)

▷ SinglePieceMagmaWithObjects(magma, obs) (operation)

▷ ObjectList(mwo) (attribute)

▷ RootObject(mwo) (attribute)

The simplest construction for a magma with objects M is to take a magma m and an ordered set s, and form arrows (u, a, v) for every a in m and u, v in s. Multiplication is defined by (u, a, v) \* (v, b, w) = (u, a \* b, w). In this package we prefer to write (u, a, v) as  $(a : u \rightarrow v)$ , so that the multiplication rule becomes  $(a : u \rightarrow v) * (b : v \rightarrow w) = (a * b : u \rightarrow w)$ .

Any finite, ordered set is in principle acceptable as the object list of M, but most of the time we find it convenient to restrict ourselves to sets of non-positive integers.

This is the only construction implemented here for magmas, semigroups, and monoids with objects, and these all have the property IsDirectProductWithCompleteDigraph. There are other constructions implemented for groupoids.

The *root object* of *M* is the first element in *s*.

```
_{-} Example _{-}
gap> tm := [[1,2,4,3],[1,2,4,3],[3,4,2,1],[3,4,2,1]];;
gap> Display( tm );
[[1, 2, 4, 3],
    1, 2, 4,
                3],
               1],
  [ 3, 4, 2,
  [ 3, 4, 2, 1]]
gap> m := MagmaByMultiplicationTable( tm );; SetName( m, "m" );
gap> m1 := MagmaElement(m,1);; m2 := MagmaElement(m,2);;
gap> m3 := MagmaElement(m,3);; m4 := MagmaElement(m,4);;
gap> M78 := MagmaWithObjects( m, [-8,-7] );
magma with objects :-
   magma = m
 objects = [-8, -7]
gap> SetName( M78, "M78" );
gap> [ IsAssociative(M78), IsCommutative(M78) ];
[false, false]
gap> [ RootObject( M78 ), ObjectList( M78 ) ];
[ -8, [ -8, -7 ] ]
```

## 2.1.2 IsDomainWithObjects

The output from function MagmaWithObjects lies in the categories IsDomainWithObjects, IsMagmaWithObjects and CategoryCollections(IsMultiplicativeElementWithObjects). As composition is only partial, the output does *not* lie in the category IsMagma.

```
gap> [ IsDomainWithObjects(M78), IsMagmaWithObjects(M78), IsMagma(M78) ];
[ true, true, false ]
```

#### **2.1.3** Arrow

```
▷ Arrow(mwo, elt, tail, head) (operation)
▷ ElementOfArrow(arr) (operation)
```

Arrows in a magma with objects lie in the category IsMultiplicativeElementWithObjects. An attempt to multiply two arrows which do not compose resuts in fail being returned. Each arrow  $arr=(a:u\to v)$  has three components. The magma  $element\ a\in m$  may be accessed by ElementOfArrow(arr). Similarly, the tail object u and the head object v may be obtained using TailOfArrow(arr) and HeadOfArrow(arr) respectively. The operation MultiplicativeElementWithObjects is a synonym for Arrow since this was used in older versions of the package.

```
gap> a78 := Arrow( M78, m2, -7, -8 );
[m2 : -7 -> -8]
gap> a78 in M78;
true
gap> b87 := Arrow( M78, m4, -8, -7 );;
gap> [ ElementOfArrow( b87 ), TailOfArrow( b87 ), HeadOfArrow( b87 ) ];
[ m4, -8, -7 ]
gap> ba := b87*a78;; ab := a78*b87;; [ ba, ab ];
[ [m4 : -8 -> -8], [m3 : -7 -> -7] ]
gap> [ a78^2, ba^2, ba^3 ];
[ fail, [m1 : -8 -> -8], [m3 : -8 -> -8] ]
gap> ## this demonstrates non-associativity:
gap> [ a78*ba, ab*a78, a78*ba=ab*a78 ];
[ [m3 : -7 -> -8], [m4 : -7 -> -8], false ]
```

#### 2.1.4 IsSinglePieceDomain

If the partial composition is forgotten, then what remains is a digraph (usually with multiple edges and loops). Thus the notion of *connected component* may be inherited by magmas with objects from digraphs. Unfortunately the terms Component and Constituent are already in considerable use elsewhere in GAP, so (and this may change if a more suitable word is suggested) we use the term IsSinglePieceDomain to describe an object with an underlying connected digraph. The property IsSinglePiece is a synonym for IsSinglePieceDomain and IsMagmaWithObjects. When each connected component has a single object, and there is more than one component, the magma with objects is *discrete*.

```
gap> IsSinglePiece( M78 );
true
gap> IsDirectProductWithCompleteDigraph( M78 );
true
```

```
gap> IsDiscreteMagmaWithObjects( M78 );
false
```

## 2.2 Semigroups with objects

## 2.2.1 SemigroupWithObjects

```
▷ SemigroupWithObjects(args) (function)

▷ SinglePieceSemigroupWithObjects(sgp, obs) (operation)

▷ DomainWithSingleObject(dom, obj) (operation)
```

The constructions in section 2.1 give a SinglePieceSemigroupWithObjects when the magma is a semigroup. In the example we use a transformation semigroup and 3 objects.

```
Example
gap> t := Transformation( [1,1,2,3] );;
gap> s := Transformation( [2,2,3,3] );;
gap> r := Transformation( [2,3,4,4] );;
gap> sgp := Semigroup( t, s, r );;
gap> SetName( sgp, "sgp<t,s,r>" );
gap> S123 := SemigroupWithObjects( sgp, [-3,-2,-1] );
semigroup with objects :-
    magma = sgp<t,s,r>
  objects = [ -3, -2, -1 ]
gap> [ IsAssociative(S123), IsCommutative(S123) ];
[ true, false ]
gap> t12 := Arrow( S123, t, -1, -2 );
[Transformation( [ 1, 1, 2, 3 ] ) : -1 \rightarrow -2]
gap> s23 := Arrow( S123, s, -2, -3 );
[Transformation( [ 2, 2, 3, 3 ] ) : -2 \rightarrow -3]
gap> r31 := Arrow( S123, r, -3, -1 );
[Transformation( [ 2, 3, 4, 4 ] ) : -3 \rightarrow -1]
gap > ts13 := t12 * s23;
[Transformation( [ 2, 2, 2, 3 ] ) : -1 \rightarrow -3]
gap> sr21 := s23 * r31;
[Transformation([3, 3, 4, 4]): -2 -> -1]
gap> rt32 := r31 * t12;
[Transformation( [ 1, 2, 3, 3 ] ) : -3 \rightarrow -2]
gap> tsr1 := ts13 * r31;
[Transformation([3, 3, 3]): -1 \rightarrow -1]
```

A magma, semigroup, monoid, or group can be made into a magma with objects by the addition of a single object. The two are algebraically isomorphic, and there is one arrow (a loop) for each element in *dom*. In the example we take the semigroup sgp of size 17 at the object 0.

```
gap> S0 := DomainWithSingleObject( sgp, 0 );
semigroup with objects :-
```

```
magma = sgp<t,s,r>
objects = [ 0 ]
gap> t0 := Arrow( S0, t, 0, 0 );
[Transformation( [ 1, 1, 2, 3 ] ) : 0 -> 0]
gap> Size( S0 );
17
```

## 2.3 Monoids with objects

## 2.3.1 MonoidWithObjects

```
▷ MonoidWithObjects(args) (function)
▷ SinglePieceMonoidWithObjects(mon, obs) (operation)
```

The constructions in section 2.1 give a SinglePieceMonoidWithObjects when the magma is a monoid. The example uses a finitely presented monoid with 2 generators and 2 objects.

```
____ Example
gap> fm := FreeMonoid( 2, "f" );;
gap> em := One( fm );;
gap> gm := GeneratorsOfMonoid( fm );;
gap > mon := fm/[[gm[1]^3,em], [gm[1]*gm[2],gm[2]]];;
gap> M49 := MonoidWithObjects( mon, [-9,-4] );
monoid with objects :-
   magma = Monoid( [ f1, f2 ] )
  objects = [-9, -4]
gap> ktpo := KnownTruePropertiesOfObject( M49 );
[ "IsDuplicateFree", "IsAssociative", "IsSinglePieceDomain",
  "IsDirectProductWithCompleteDigraphDomain" ]
gap> catobj := CategoriesOfObject( M49 );;
[ "IsListOrCollection", "IsCollection", "IsExtLElement",
  "CategoryCollections(IsExtLElement)", "IsExtRElement",
  "CategoryCollections(IsExtRElement)",
  "CategoryCollections(IsMultiplicativeElement)", "IsGeneralizedDomain",
  "IsDomainWithObjects",
  "CategoryCollections(IsMultiplicativeElementWithObjects)",
  "CategoryCollections(IsMultiplicativeElementWithObjectsAndOnes)",
  "CategoryCollections(IsMultiplicativeElementWithObjectsAndInverses)",
  "IsMagmaWithObjects", "IsSemigroupWithObjects", "IsMonoidWithObjects"]
```

## 2.4 Generators of magmas with objects

## 2.4.1 GeneratorsOfMagmaWithObjects

```
▷ GeneratorsOfMagmaWithObjects(mwo) (operation)
▷ GeneratorsOfSemigroupWithObjects(swo) (operation)
```

□ GeneratorsOfMonoidWithObjects(mwo)

(operation)

For a magma or semigroup with objects, the generating set consists of arrows  $(g: u \rightarrow v)$  for every pair of objects u, v and every generating element for the magma or semigroup.

For a monoid with objects, the generating set consists of two parts. Firstly, there is a loop at the root object r for each generator of the monoid. Secondly, for each object u distinct from r, there are arrows  $(1:r \to u)$  and  $(1:u \to r)$ . (Perhaps only one of each pair is required?) Then

```
(e: u \to v) = (1: u \to r) * (e: r \to r) * (1: r \to v).
```

```
_____ Example ____
gap> GeneratorsOfMagmaWithObjects( M78 );
[ [m1 : -8 -> -8], [m2 : -8 -> -8], [m3 : -8 -> -8], [m4 : -8 -> -8],
  [m1 : -8 -> -7], [m2 : -8 -> -7], [m3 : -8 -> -7], [m4 : -8 -> -7],
  [m1 : -7 -> -8], [m2 : -7 -> -8], [m3 : -7 -> -8], [m4 : -7 -> -8],
  [m1 : -7 -> -7], [m2 : -7 -> -7], [m3 : -7 -> -7], [m4 : -7 -> -7]]
gap> genS := GeneratorsOfSemigroupWithObjects( S123 );;
gap> Length( genS );
27
gap> genM := GeneratorsOfMonoidWithObjects( M49 );
[ [f1 : -9 \rightarrow -9], [f2 : -9 \rightarrow -9], [<identity ...> : -9 -> -4],
  [<identity ...> : -4 -> -9]
gap> g1:=genM[2];; g2:=genM[3];; g3:=genM[4];; g4:=genM[5];;
gap> [g4,g2,g1,g3];
[ (\text{identity } ... > : -4 -> -9], [f2 : -9 -> -9], [f1 : -9 -> -9],
  [<identity ...> : -9 -> -4] ]
gap> g4*g2*g1*g3;
[f2*f1 : -4 -> -4]
```

## 2.5 Structures with more than one piece

#### 2.5.1 UnionOfPieces (for magmas with objects)

```
▷ UnionOfPieces(pieces) (operation)
▷ Pieces(mwo) (attribute)
▷ PieceOfObject(mwo, obj) (operation)
```

A magma with objects whose underlying digraph has two or more connected components can be constructed by taking the union of two or more connected structures. These, in turn, can be combined together. The only requirement is that all the object lists should be disjoint. The pieces are ordered by the order of their root objects.

```
gap> N1 := UnionOfPieces( [ M78, S123 ] );
magma with objects having 2 pieces :-
1: M78
2: semigroup with objects :-
magma = sgp<t,s,r>
```

```
objects = [ -3, -2, -1 ]
gap> ObjectList( N1 );
[-8, -7, -3, -2, -1]
gap> Pieces(N1);
[ M78, semigroup with objects :-
       magma = sgp<t,s,r>
      objects = [-3, -2, -1]
gap> PieceOfObject( N1, -7 );
gap> N2 := UnionOfPieces( [ M49, S0 ] );
semigroup with objects having 2 pieces :-
1: monoid with objects :-
   magma = Monoid( [ f1, f2 ] )
 objects = [-9, -4]
2: semigroup with objects :-
   magma = sgp<t,s,r>
  objects = [ 0 ]
gap> ObjectList( N2 );
[ -9, -4, 0 ]
gap> N3 := UnionOfPieces( [ N1, N2] );
magma with objects having 4 pieces :-
1: monoid with objects :-
   magma = Monoid( [ f1, f2 ] )
  objects = [-9, -4]
2: M78
3: semigroup with objects :-
   magma = sgp<t,s,r>
 objects = [ -3, -2, -1 ]
4: semigroup with objects :-
   magma = sgp<t,s,r>
  objects = [ 0 ]
gap> ObjectList( N3 );
[-9, -8, -7, -4, -3, -2, -1, 0]
gap> Length( GeneratorsOfMagmaWithObjects( N3 ) );
gap> ## the next command returns fail since the object sets are not disjoint:
gap> N4 := UnionOfPieces( [ S123, DomainWithSingleObject( sgp, -2 ) ] );
fail
```

# **Chapter 3**

# **Mappings of many-object structures**

A homomorphism f from a magma with objects M to a magma with objects N consists of

- a map  $f_O$  from the objects of M to those of N,
- a map  $f_A$  from the arrows of M to those of N.

The map  $f_A$  is required to be compatible with the tail and head maps and to preserve multiplication:

$$f_A(a:u\to v)*f_A(b:v\to w) = f_A(a*b:u\to w)$$

with tail  $f_O(u)$  and head  $f_O(w)$ .

When the underlying magma of M is a monoid or group, the map  $f_A$  is required to preserve identities and inverses.

## 3.1 Homomorphisms of magmas with objects

#### 3.1.1 MagmaWithObjectsHomomorphism

▷ MagmaWithObjectsHomomorphism(args) (function)
▷ HomomorphismFromSinglePiece(src, rng, hom, imobs) (operation)
▷ HomomorphismToSinglePiece(src, rng, images) (operation)
▷ MappingToSinglePieceData(mwohom) (attribute)
▷ PiecesOfMapping(mwohom) (attribute)
▷ IsomorphismNewObjects(src, objlist) (operation)

There are a variety of homomorphism constructors.

The simplest construction gives a homomorphism  $M \to N$  with both M and N connected. It is implemented as IsMappingToSinglePieceRep with attributes Source, Range and MappingToSinglePieceData. The operation requires the following information:

- a magma homomorphism hom from the underlying magma of M to the underlying magma of N,
- a list imobs of the images of the objects of M.

In the first example we construct endomappings of m and M78.

```
Example
gap> tup1 := [ DirectProductElement([m1,m2]), DirectProductElement([m2,m1]),
               DirectProductElement([m3,m4]), DirectProductElement([m4,m3]) ];;
gap> f1 := GeneralMappingByElements( m, m, tup1 );
gap> IsMagmaHomomorphism( f1 );
true
gap> hom1 := MagmaWithObjectsHomomorphism( M78, M78, f1, [-8,-7] );;
gap> Display( hom1 );
homomorphism to single piece magma: M78 -> M78
magma hom: <mapping: m -> m >
object map: [ -8, -7 ] -> [ -8, -7 ]
gap> [ Source( hom1 ), Range( hom1 ) ];
[ M78, M78 ]
gap> b87;
[m4 : -8 -> -7]
gap> im1 := ImageElm( hom1, b87 );
[m3 : -8 -> -7]
gap> i56 := IsomorphismNewObjects( M78, [-5,-6] );
magma with objects homomorphism :
[ [ IdentityMapping( m ), [ -5, -6 ] ] ]
gap> ib87 := ImageElm( i56, b87 );
[m4 : -5 -> -6]
gap> M65 := Range( i56);;
gap> SetName( M65, "M65" );
gap> j56 := InverseGeneralMapping( i56 );;
gap> ImagesOfObjects( j56 );
[-7, -8]
gap > comp := j56 * hom1;
magma with objects homomorphism : M65 -> M78
[ [ <mapping: m -> m >, [ -7, -8 ] ] ]
gap> ImageElm( comp, ib87 );
[m3 : -8 -> -7]
```

A homomorphism *to* a connected magma with objects may have a source with several pieces, and so is a union of homomorphisms *from* single pieces.

```
gap> M4 := UnionOfPieces( [ M78, M65 ] );;
gap> images := [ MappingToSinglePieceData( hom1 )[1],
> MappingToSinglePieceData( j56 )[1] ];
[ [ <mapping: m -> m >, [ -8, -7 ] ], [ IdentityMapping( m ), [ -7, -8 ] ] ]
gap> map4 := HomomorphismToSinglePiece( M4, M78, images );
magma with objects homomorphism :
[ [ <mapping: m -> m >, [ -8, -7 ] ], [ IdentityMapping( m ), [ -7, -8 ] ] ]
gap> ImageElm( map4, b87 );
[m3 : -8 -> -7]
gap> ImageElm( map4, ib87 );
[m4 : -8 -> -7]
```

## 3.2 Homomorphisms of semigroups and monoids with objects

The next example exhibits a homomorphism between transformation semigroups with objects.

```
Example
gap> t2 := Transformation( [2,2,4,1] );;
gap> s2 := Transformation([1,1,4,4]);;
gap> r2 := Transformation( [4,1,3,3] );;
gap> sgp2 := Semigroup( [ t2, s2, r2 ] );;
gap> SetName( sgp2, "sgp<t2,s2,r2>" );
gap> ## apparently no method for transformation semigroups available for:
gap> ## nat := NaturalHomomorphismByGenerators( sgp, sgp2 ); so we use:
gap> ## in the function flip below t is a transformation on [1..n]
gap> flip := function( t )
     local i, j, k, L, L2, n;
     n := DegreeOfTransformation( t );
      L := ImageListOfTransformation( t );
      if IsOddInt(n) then n:=n+1; L1:=Concatenation(L,[n]);
                     else L1:=L; fi;
>
     L2 := ShallowCopy( L1 );
>
      for i in [1..n] do
>
          if IsOddInt(i) then j:=i+1; else j:=i-1; fi;
         k := L1[i];
>
          if IsOddInt(k) then L2[i]:=k+1; else L2[i]:=k-1; fi;
>
      od:
      return( Transformation( L2 ) );
> end;;
gap> smap := MappingByFunction( sgp, sgp2, flip );;
gap> ok := RespectsMultiplication( smap );
gap> [ t, ImageElm( smap, t ) ];
[ Transformation([1, 1, 2, 3]), Transformation([2, 2, 4, 1])]
gap> [ s, ImageElm( smap, s ) ];
[ Transformation( [ 2, 2, 3, 3 ] ), Transformation( [ 1, 1, 4, 4 ] ) ]
gap> [ r, ImageElm( smap, r ) ];
[ Transformation( [ 2, 3, 4, 4 ] ), Transformation( [ 4, 1, 3, 3 ] ) ]
gap> SetName( smap, "smap" );
gap> T123 := SemigroupWithObjects( sgp2, [-13,-12,-11] );;
gap> shom := MagmaWithObjectsHomomorphism( S123, T123, smap, [-11,-12,-13] );;
gap> it12 := ImageElm( shom, t12 );; [ t12, it12 ];
[ [Transformation( [ 1, 1, 2, 3 ] ) : -1 \rightarrow -2],
  [Transformation([2, 2, 4, 1]): -13 -> -12]
gap> is23 := ImageElm( shom, s23 );; [ s23, is23 ];
[ [Transformation( [ 2, 2, 3, 3 ] ) : -2 -> -3],
  [Transformation([1, 1, 4, 4]): -12 -> -11]]
gap> ir31 := ImageElm( shom, r31 );; [ r31, ir31 ];
[ [Transformation( [ 2, 3, 4, 4 ] ) : -3 -> -1],
  [Transformation([4, 1, 3, 3]): -11 -> -13]]
```

## 3.3 Homomorphisms to more than one piece

## 3.3.1 HomomorphismByUnion (for magmas with objects)

When  $f: M \to N$  and N has more than one connected component, then f is a union of homomorphisms, one for each piece in the range.

```
_ Example _
gap> N4 := UnionOfPieces( [ M78, T123 ] );
magma with objects having 2 pieces :-
1: semigroup with objects :-
   magma = sgp < t2, s2, r2 >
  objects = [-13, -12, -11]
2: M78
gap> h14 := HomomorphismByUnionNC( N1, N4, [ hom1, shom ] );
magma with objects homomorphism :
[ magma with objects homomorphism : M78 -> M78
    [ [ <mapping: m -> m >, [ -8, -7 ] ] ], magma with objects homomorphism :
    [ [ smap, [ -11, -12, -13 ] ] ]
gap> ImageElm( h14, a78 );
[m1 : -7 -> -8]
gap> ImageElm( h14, r31 );
[Transformation([4, 1, 3, 3]): -11 -> -13]
```

## 3.3.2 IsInjectiveOnObjects

The meaning of these five properties is obvious.

```
gap> IsInjectiveOnObjects( h14 );
true
gap> IsSurjectiveOnObjects( h14 );
true
gap> IsBijectiveOnObjects( h14 );
true
gap> IsEndomorphismWithObjects( h14 );
false
gap> IsAutomorphismWithObjects( h14 );
false
```

## 3.4 Mappings defined by a function

## 3.4.1 MappingWithObjectsByFunction

```
▷ MappingWithObjectsByFunction(src, rng, fun, imobs) (operation)
▷ IsMappingWithObjectsByFunction(map) (property)
▷ UnderlyingFunction(map) (attribute)
```

More general mappings, which need not preserve multiplication, are available using this operation. See section 5.6 for an application.

```
gap> flip := function(a) return Arrow(M78,a![1],a![3],a![2]); end;
function(a) ... end
gap> flipmap := MappingWithObjectsByFunction( M78, M78, flip, [-8,-7]);
magma with objects mapping by function : M78 -> M78
function: function (a)
    return Arrow( M78, a![1], a![3], a![2]);
end
gap> a78; ImageElm( flipmap, a78);
[m2 : -7 -> -8]
[m2 : -8 -> -7]
```

## Chapter 4

# **Groupoids**

A *groupoid* is a (mathematical) category in which every element is invertible. It consists of a set of *pieces*, each of which is a connected groupoid. The usual terminology is 'connected component', but in GAP 'component' is used for 'record component', so we use the term *single piece*.

The simplest form for a *single piece groupoid* is the direct product of a group and a complete digraph, and so is determined by a set of *objects* obs =  $\Omega$  (the least of which is the *root object*), and a *root group* grp = G. Then the elements of the groupoid are *arrows*  $g: o_1 \rightarrow o_2$ , stored as triples [g,o1,o2], where  $g \in G$  and  $o_1,o_2 \in \Omega$ . The objects will generally be chosen to be consecutive negative integers, but any suitable ordered set is acceptable, and 'consecutive' is not a requirement. The root group will usually be taken to be a permutation group, but pc-groups, fp-groups and matrix groups are also supported.

A group may be considered as a single piece groupoid with one object.

A *groupoid* is a set of one or more single piece groupoids, its *pieces*, and is represented as IsGroupoidRep, with attribute PiecesOfGroupoid.

The underlying digraph of a single piece groupoid is a regular, complete digraph on the object set  $\Omega$  with |G| arrows from any one object to any other object. It will be convenient to specify a set of rays consisting of  $|\Omega|$  arrows  $r_i: o_1 \to o_i$ , where  $o_1$  is the root object and  $r_1$  is the identity in G. In the simplest examples all the  $r_i$  will be identity elements, but rays are useful when forming subgroupoids (see SubgroupoidWithRays (4.3.2)).

A groupoid is *homogeneous* if it has two of more isomorphic pieces, with identical groups. The special case of *homogeneous*, *discrete* groupoids, where each piece has a single object, is given its own representation. These are used in the XMod package as the source of a crossed modules of groupoids.

For the definitions of the standard properties of groupoids we refer to R. Brown's book "Topology" [Bro88], recently revised and reissued as "Topology and Groupoids" [Bro06].

## 4.1 Groupoids: their properties and attributes

#### 4.1.1 SinglePieceGroupoid

```
▷ SinglePieceGroupoid(grp, obs) (operation)

▷ Groupoid(args) (function)

▷ DomainWithSingleObject(gp, obj) (operation)
```

The simplest construction of a groupoid is as the direct product of a group and a complete digraph.

Such a groupoid will be called a *standard groupoid*. Many subgroupoids of such a groupoid do not have this simple form, and will be considered in section 4.3. The global function Groupoid will normally find the appropriate constructor to call, the options being:

- the object group, a set of objects;
- a group being converted to a groupoid, a single object;
- a list of groupoids which have already been constructed (see 4.1.4).

Methods for ViewObj, PrintObj and Display are provided for groupoids and the other types of object in this package. Users are advised to supply names for all the groups and groupoids they construct.

```
gap> s4 := Group( (1,2,3,4), (3,4) );;
gap> d8 := Subgroup( s4, [ (1,2,3,4), (1,3) ] );;
gap> SetName( s4, "s4" ); SetName( d8, "d8" );
gap> Gs4 := SinglePieceGroupoid( s4, [-15 .. -11] );
single piece groupoid: < s4, [ -15 .. -11 ] >
gap> Gd8 := Groupoid( d8, [-9,-8,-7] );
single piece groupoid: < d8, [ -9, -8, -7 ] >
gap> c6 := Group( (5,6,7)(8,9) );;
gap> SetName( c6, "c6" );
gap> Gc6 := DomainWithSingleObject( c6, -6 );
single piece groupoid: < c6, [ -6 ] >
gap> SetName( Gs4, "Gs4" ); SetName( Gd8, "Gd8" ); SetName( Gc6, "Gc6" );
```

## 4.1.2 ObjectList (for groupoids)

The ObjectList of a groupoid is the sorted list of its objects. The RootObject in a single-piece groupoid is the object with the least label. A *loop* is an arrow of the form  $g:o\to o$ , and the loops at a particular object o form a group, the ObjectGroup at o. The RootGroup is the ObjectGroup at the RootObject.

In the example, the groupoids Gf2c6 and Gabc illustrate that the objects need not be integers.

```
gap> ObjectList( Gs4 );
[ -15 .. -11 ]
gap> f2 := FreeGroup(2);;
gap> Gf2c6 := Groupoid( c6, GeneratorsOfGroup(f2) );
single piece groupoid: < c6, [ f1, f2 ] >
gap> Arrow( Gf2c6, (5,7,6), f2.1, f2.2 );
[(5,7,6) : f1 -> f2]
gap> Gabc := Groupoid( d8, [ "a", "b", "c" ] );
```

```
single piece groupoid: < d8, [ "a", "b", "c" ] >
gap> Arrow( Gabc, (2,4), "c", "b" );
[(2,4) : c -> b]
```

## 4.1.3 IsPermGroupoid

```
▷ IsPermGroupoid(gpd) (property)

▷ IsPcGroupoid(gpd) (property)

▷ IsFpGroupoid(gpd) (property)

▷ IsMatrixGroupoid(gpd) (property)

▷ IsFreeGroupoid(gpd) (property)
```

A groupoid is a permutation groupoid if all its pieces have permutation groups. Most of the examples in this chapter are permutation groupoids, but in principle any type of group known to GAP may be used.

In the following example Gf2 is an fp-groupoid and also a free groupoid, Gq8 is a pc-groupoid, and Gg143 is a matrix groupoid. See section 5.7 for matrix representations of groupoids.

## **4.1.4** UnionOfPieces (for groupoids)

```
    ▷ UnionOfPieces(pieces) (operation)
    ▷ Pieces(gpd) (attribute)
    ▷ Size(gpd) (attribute)
    ▷ ReplaceOnePieceInUnion(U, old_piece, new_piece) (operation)
```

When a groupoid consists of two or more pieces, we require their object lists to be disjoint. The operation UnionOfPieces and the attribute Pieces, introduced in section 2.5, are also used for groupoids. The pieces are sorted by the least object in their object lists. The ObjectList is the sorted concatenation of the objects in the pieces.

The Size of a groupoid is the number of its arrows. For a single piece groupoid, this is the product of the size of the group with the square of the number of objects. For a non-connected groupoid, the size is the sum of the sizes of its pieces.

One of the pieces in a groupoid may be replaced by an alternative piece using the operation ReplaceOnePieceInUnion. The *old\_piece* may be either the *position* of the piece to be replaced, or one of the pieces in U. The objects in the new piece may or may not overlap the objects in the piece being removed – we just require that the object lists in the new union are disjoint.

```
_ Example
gap> U3 := UnionOfPieces( [ Gc6, Gd8, Gs4 ] );;
gap> Display( U3 );
groupoid with 3 pieces:
< objects: [ -15 .. -11 ]</pre>
    group: s4 = \langle (1,2,3,4), (3,4) \rangle \rangle
< objects: [ -9, -8, -7 ]</pre>
    group: d8 = \langle (1,2,3,4), (1,3) \rangle \rangle
< objects: [ -6 ]
    group: c6 = \langle (5,6,7)(8,9) \rangle \rangle
gap> Pieces( U3 );
[ Gs4, Gd8, Gc6 ]
gap> ObjectList( U3 );
[-15, -14, -13, -12, -11, -9, -8, -7, -6]
gap> U2 := Groupoid( [ Gf2, Gq8 ] );;
gap> [ Size(Gs4), Size(Gd8), Size(Gc6), Size(U3) ];
[ 600, 72, 6, 678 ]
gap> [ Size(Gf2), Size(Gq8), Size(U2) ];
[ infinity, 32, infinity ]
gap> U5 := UnionOfPieces( [ U3, U2 ] );
groupoid with 5 pieces:
[ Gf2, Gq8, Gs4, Gd8, Gc6 ]
gap> Display( U5 );
groupoid with 5 pieces:
< objects: [ -22 ]</pre>
    group: f2 = <[ f1, f2 ]> >
< objects: [ -18, -17 ]
    group: q8 = \langle [x, y, y2] \rangle \rangle
< objects: [ -15 .. -11 ]
    group: s4 = \langle (1,2,3,4), (3,4) \rangle \rangle
< objects: [ -9, -8, -7 ]</pre>
    group: d8 = \langle (1,2,3,4), (1,3) \rangle \rangle
< objects: [ -6 ]</pre>
    group: c6 = \langle (5,6,7)(8,9) \rangle \rangle
gap> ## in the next example the object lists are not disjoint
gap> UnionOfPieces( [ Gf2, Gg143 ] );
gap> V3 := ReplaceOnePieceInUnion( U3, Gd8, Gq8 );
groupoid with 3 pieces:
[ Gq8, Gs4, Gc6 ]
gap> ObjectList( V3 );
[ -18, -17, -15, -14, -13, -12, -11, -6 ]
gap> U3 = V3;
gap> V2 := ReplaceOnePieceInUnion( U2, 2, Gd8 );
groupoid with 2 pieces:
[ Gf2, Gd8 ]
```

#### 4.1.5 HomogeneousGroupoid

```
▷ HomogeneousGroupoid(gpd, oblist) (operation)
▷ HomogeneousDiscreteGroupoid(gp, obs) (operation)
▷ PieceIsomorphisms(hgpd) (attribute)
```

Special functions are provided for the case where a groupoid has more than one connected component, and when these components are identical except for their object sets. Such groupoids are said to be *homogeneous*.

The operation HomogeneousGroupoid is used when the components each contain more than one object. The arguments consist of a single piece groupoid gpd and a list of lists of objects oblist, each of whose lists has the same length as the object list obs of gpd. Note that gpd is *not* included as one of the pieces in the output unless obs is included as one of the lists in oblist.

The operation HomogeneousDiscreteGroupoid is used when the components each have a single object. In this case the first argument is just a group – the root group for each component. These groupoids are used in the XMod package as the source of many crossed modules of groupoids.

Both types of groupoid have the property IsHomogeneousDomainWithObjects. In the latter case a separate representation IsHomogeneousDiscreteGroupoidRep is used.

```
Example
gap> Hd8 := HomogeneousGroupoid( Gd8,
                 [[-20,-19,-18], [-12,-11,-10], [-16,-15,-14]]);
homogeneous groupoid with 3 pieces:
1: single piece groupoid: < d8, [ -20, -19, -18 ] >
2: single piece groupoid: < d8, [ -16, -15, -14 ] >
3: single piece groupoid: < d8, [ -12, -11, -10 ] >
gap> Size(Hd8);
                   ## 8x3x3 + 8x3x3
216
gap> IsHomogeneousDomainWithObjects( Hd8 );
gap> PieceIsomorphisms( Hd8 );
[ groupoid homomorphism :
    [[(1,2,3,4): -20 \rightarrow -20], [(1,3): -20 \rightarrow -20], [(): -20 \rightarrow -19],
           [(): -20 \rightarrow -18]],
       [ [(1,2,3,4) : -16 \rightarrow -16], [(1,3) : -16 \rightarrow -16], [() : -16 \rightarrow -15],
           [() : -16 \rightarrow -14] ] ], groupoid homomorphism :
    [[(1,2,3,4): -20 \rightarrow -20], [(1,3): -20 \rightarrow -20], [(): -20 \rightarrow -19],
           [(): -20 \rightarrow -18]],
      [(1,2,3,4): -12 \rightarrow -12], [(1,3): -12 \rightarrow -12], [(): -12 \rightarrow -11],
           [(): -12 -> -10]]]]
gap> Hc6 := HomogeneousDiscreteGroupoid( c6, [-7..-4] );
homogeneous, discrete groupoid: < c6, [ -7 .. -4 ] >
gap> Size( Hc6 );
                     ## 6x4
gap> RepresentationsOfObject( Gd8 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsMWOSinglePieceRep" ]
gap> RepresentationsOfObject( Hd8 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPiecesRep" ]
```

```
gap> RepresentationsOfObject( Hc6 );
[ "IsComponentObjectRep", "IsAttributeStoringRep",
    "IsHomogeneousDiscreteGroupoidRep" ]
gap> ktpo := KnownTruePropertiesOfObject( Hc6 );;
gap> ans :=
> [ "IsDuplicateFree", "IsAssociative", "IsCommutative",
>    "IsDiscreteDomainWithObjects", "IsHomogeneousDomainWithObjects" ];;
gap> ForAll( ans, a -> ( a in ktpo ) );
true
```

## 4.1.6 DirectProductOp

```
▷ DirectProductOp(list, gpd) (operation)
▷ Projection(gpd, pos) (operation)
▷ Embedding(gpd, pos) (operation)
```

The direct product of groupoids G,H has as group the direct product of the groups in G and H and as object list the cartesian product of their object lists. As usual with DirectProductOp the two parameters are a list of groupoids followed by the first entry in the list.

```
Example
gap> prod := DirectProductOp( [Gd8,Gc6], Gd8 );
single piece groupoid: < Group([(1,2,3,4), (1,3), (5,6,7)(8,9)]),
[[-9, -6], [-8, -6], [-7, -6]] >
gap> Projection( prod, 1 );
groupoid homomorphism :
[[(1,2,3,4):[-9,-6]->[-9,-6]],[(1,3):[-9,-6]->[-9,-6]]
        , [(5,6,7)(8,9) : [ -9, -6 ] -> [ -9, -6 ]],
      [():[-9,-6]\rightarrow[-8,-6]],[():[-9,-6]\rightarrow[-7,-6]]],
  [ [(1,2,3,4) : -9 \rightarrow -9], [(1,3) : -9 \rightarrow -9], [() : -9 \rightarrow -9],
      [() : -9 \rightarrow -8], [() : -9 \rightarrow -7]]
gap> Embedding( prod, 2 );
groupoid homomorphism :
[[[(5,6,7)(8,9):-6 \rightarrow -6]],
  [[(5,6,7)(8,9) : [-9, -6] -> [-9, -6]]]
gap> DirectProductInfo( prod );
rec( embeddings := [ , groupoid homomorphism :
        [[(5,6,7)(8,9):-6->-6]],
          [[(5,6,7)(8,9):[-9,-6]-9,-6]]]]], first := Gd8,
  groupoids := [ Gd8, Gc6 ], groups := [ d8, c6 ],
  objectlists := [ [ -9, -8, -7 ], [ -6 ] ],
  projections := [ groupoid homomorphism :
        [[[(1,2,3,4):[-9,-6]->[-9,-6]],
              [(1,3) : [-9, -6] \rightarrow [-9, -6]],
              [(5,6,7)(8,9) : [-9,-6] \rightarrow [-9,-6]],
              [() : [-9, -6] \rightarrow [-8, -6]],
              [() : [-9, -6] \rightarrow [-7, -6]]],
          [[(1,2,3,4): -9 \rightarrow -9], [(1,3): -9 \rightarrow -9], [(): -9 \rightarrow -9],
              [(): -9 -> -8], [(): -9 -> -7]]]
```

## 4.2 Groupoid elements; stars; costars; homsets

#### 4.2.1 GroupoidElement

```
▷ GroupoidElement(gpd, elt, tail, head) (operation)
▷ ElementOfArrow(elt) (operation)
▷ TailOfArrow(elt) (operation)
▷ HeadOfArrow(elt) (operation)
```

The operation GroupoidElement is a synonym for the operation Arrow, as described in subsection 2.1.3. To recapitulate, an arrow e consists of a group element, ElementOfArrow(e); the tail (source) object, TailOfArrow(e); and the head (target) object, HeadOfArrow(e). Arrows have a partial composition: two arrows may be multiplied when the head of the first coincides with the tail of the second. If an attempt is made to multiply arrows where this condition does not hold, then the value fail is returned.

```
\_ Example _{-}
gap> e1 := GroupoidElement( Gd8, (1,2,3,4), -9, -8 );
[(1,2,3,4): -9 \rightarrow -8]
gap> e2 := Arrow( Gd8, (1,3), -8, -7 );
[(1,3) : -8 \rightarrow -7]
gap> Print( [ ElementOfArrow(e1), TailOfArrow(e1), HeadOfArrow(e1) ], "\n" );
[ (1,2,3,4), -9, -8 ]
gap> e1e2 := e1*e2;
[(1,2)(3,4): -9 \rightarrow -7]
gap> e2*e1;
fail
gap> e3 := Arrow( Gd8, (2,4), -7, -9 );;
gap> loop := e1e2*e3;
[(1,4,3,2): -9 \rightarrow -9]
gap> loop^2;
[(1,3)(2,4): -9 \rightarrow -9]
```

## 4.2.2 IdentityArrow

```
\triangleright IdentityArrow(gpd, obj) (operation)
```

The identity arrow  $1_o$  of G at object o is  $(e: o \to o)$  where e is the identity element in the object group. The *inverse arrow*  $e^{-1}$  of  $e = (c: p \to q)$  is  $(c^{-1}: q \to p)$ , so that  $e * e^{-1} = 1_p$  and  $e^{-1} * e = 1_q$ .

```
gap> i8 := IdentityArrow( Gd8, -8 );
[() : -8 -> -8]
gap> [ e1*i8, i8*e1, e1^-1];
[ [(1,2,3,4) : -9 -> -8], fail, [(1,4,3,2) : -8 -> -9] ]
```

#### **4.2.3** Order

```
▷ Order(arr) (attribute)
```

A groupoid element is a *loop* when the tail and head coincide. In this case the order of the element is defined to be the order of its group element.

```
gap> [ i8, loop ];
 [ [() : -8 -> -8], [(1,4,3,2) : -9 -> -9] ]
 gap> [ Order( i8 ), Order(loop) ];
 [ 1, 4 ]
```

#### 4.2.4 ObjectStar

```
ightharpoonup ObjectStar(gpd, obj) (operation)

ightharpoonup ObjectCostar(gpd, obj) (operation)

ightharpoonup Homset(gpd, tail, head) (operation)
```

The *star* at obj is the set of arrows which have obj as tail, while the *costar* is the set of arrows which have obj as head. The *homset* from obj1 to obj2 is the set of arrows with the specified tail and head, and so is bijective with the elements of the object groups. Thus every star and every costar is a union of homsets. The identity arrow at an object is a left identity for the star and a right identity for the costar at that object.

In order not to create unneccessarily long lists, these operations return objects of type IsHomsetCosetsRep for which an Iterator is provided. (An Enumerator is not yet implemented.)

```
_ Example __
gap> star9 := ObjectStar( Gd8, -9 );
<star at -9 with vertex group d8>
gap> Size( star9 );
gap> ## PrintSelection( star9, 19, 1 );
gap> iter := Iterator( star9 );;
gap> for i in [1..18] do a := NextIterator( iter ); od;
gap> for i in [19..24] do Print( i, " : ", NextIterator( iter ), "\n" ); od;
19 : [(1,2,3,4) : -9 \rightarrow -9]
20 : [(1,2,3,4) : -9 \rightarrow -8]
21 : [(1,2,3,4) : -9 \rightarrow -7]
22 : [(1,2)(3,4) : -9 \rightarrow -9]
23 : [(1,2)(3,4) : -9 \rightarrow -8]
24 : [(1,2)(3,4) : -9 \rightarrow -7]
gap> costar6 := ObjectCostar( Gc6, -6 );
<costar at -6 with vertex group c6>
gap> Size( costar6 );
gap> hsetq8 := Homset( Gq8, -18, -17 );
<homset -18 -> -17 with head group q8>
gap> Perform( hsetq8, Display );
```

```
[<identity> of ...: -18 -> -17]
[y2: -18 -> -17]
[y: -18 -> -17]
[y*y2: -18 -> -17]
[x: -18 -> -17]
[x: -18 -> -17]
[x*y2: -18 -> -17]
[x*y2: -18 -> -17]
[x*y: -18 -> -17]
```

## 4.3 Subgroupoids

#### 4.3.1 Subgroupoid

```
    ▷ Subgroupoid(args) (function)
    ▷ IsSubgroupoid(gpd, sgpd) (operation)
    ▷ IsWideSubgroupoid(gpd, sgpd) (operation)
```

A subgroupoid S of a groupoid G has as objects some subset of the objects of G. It is wide in G if both groupoids have the same object set. It is full if, for any two objects in S, the Homset is the same as that in G. The arrows of S are a subset of those of G, closed under multiplication and with tail and head in the chosen object set.

There are a variety of constructors for a subgroupoid of a standard groupoid, as described in for following sections. The global function Subgroupoid should call the operation appropriate to the parameters provided.

## 4.3.2 SubgroupoidWithRays

```
    SubgroupoidWithRays(gpd, sgp, rays) (operation)
    RaysOfGroupoid(gpd) (operation)
    RayArrowsOfGroupoid(gpd) (operation)
```

If groupoid G is of type IsDirectProductWithCompleteDigraph with group g and n objects, then a typical wide subgroupoid H of G is formed by choosing a subgroup h of g to be the object group at the root object q, and an arrow  $r:q\to p$  for each of the objects p. The chosen loop arrow at q must be the identity arrow. These n arrows are called the ray arrows of the subgroupoid. The arrows in the homset from p to p' have the form  $r^{-1}xr'$  where r,r' are the rays from q to p,p' respectively, and  $x\in h$ .

The operation RayArrowsOfGroupoid returns a list of arrows, one for each object, while the operation RaysOfGroupoid returns the list of group elements in these arrows.

Note that it is also possible to construct a subgroupoid with rays of a subgroupoid with rays.

In the following example we construct a subgroupoid Ua4 of the groupoid Gs4, and then a second subgroupoid Uc3. The initial standard groupoid Gs4 is set as the parent for both Ua4 and Uc3.

```
gap> a4 := Subgroup( s4, [ (1,2,3), (2,3,4) ] );;
gap> SetName( a4, "a4" );
gap> Ua4 := SubgroupoidWithRays( Gs4, a4, [(),(1,2),(2,3),(3,4),(1,4)] );
```

```
single piece groupoid with rays: < a4, [ -15 .. -11 ],
[(), (1,2), (2,3), (3,4), (1,4)] >
gap> IsSubgroupoid( Gs4, Ua4 );
gap> IsWideSubgroupoid( Gs4, Ua4 );
true
gap> RaysOfGroupoid( Ua4 );
[(), (1,2), (2,3), (3,4), (1,4)]
gap> RayArrowsOfGroupoid( Ua4 );
[ () : -15 \rightarrow -15], [(1,2) : -15 \rightarrow -14], [(2,3) : -15 \rightarrow -13],
  [(3,4): -15 \rightarrow -12], [(1,4): -15 \rightarrow -11]]
gap> c3 := Subgroup( a4, [ (1,2,3) ] );;
gap> SetName( c3, "c3" );
gap> Uc3 := SubgroupoidWithRays( Ua4, c3,
                [(), (1,2,3,4), (1,3), (2,4), (1,4,3,2)]);
single piece groupoid with rays: < c3, [ -15 .. -11 ],
[(), (1,2,3,4), (1,3), (2,4), (1,4,3,2)] >
gap> ObjectGroup( Uc3, -14 );
Group([ (2,3,4) ])
gap> Ha4 := HomogeneousGroupoid( Ua4, [ [-25..-21], [-35..-31] ] );
homogeneous groupoid with 2 pieces:
1: single piece groupoid with rays: < a4, [ -35 .. -31 ],
[(), (1,2), (2,3), (3,4), (1,4)] >
2: single piece groupoid with rays: < a4, [ -25 .. -21 ],
[(), (1,2), (2,3), (3,4), (1,4)] >
```

#### 4.3.3 SubgroupoidByObjects

```
▷ SubgroupoidByObjects(gpd, obs) (operation)
▷ SubgroupoidBySubgroup(gpd, sgp) (attribute)
```

The SubgroupoidByObjects of a groupoid gpd on a subset obs of its objects contains all the arrows of gpd with tail and head in obs.

The SubgroupoidBySubgroup of a connected groupoid gpd determinded by a subgroup sgp of the root group is the wide subgroupoid with root group sgp and containing the rays of gpd.

```
gap> Va4 := SubgroupoidByObjects( Ua4, [-14,-13,-12] );
single piece groupoid with rays: < Group( [ (1,3,2), (1,3,4) ] ),
  [ -14, -13, -12 ], [ (), (1,3,2), (1,2)(3,4) ] >
  gap> Vc3 := SubgroupoidBySubgroup( Va4, c3 );
single piece groupoid with rays: < c3, [ -14, -13, -12 ],
  [ (), (1,3,2), (1,2)(3,4) ] >
```

## 4.3.4 SubgroupoidByPieces

The most general way to construct a subgroupoid is to use the operation SubgroupoidByPieces. Its two parameters are a groupoid and a list of *pieces*, each piece being specified either as a list [sgp,obs], where sgp is a subgroup of the root group in that piece, and obs is a subset of the objects in that piece, or as a list [sgp,obs,rays] when a set of rays is required.

```
_ Example
gap> Display(Gd8);
single piece groupoid: Gd8
 objects: [ -9, -8, -7 ]
    group: d8 = \langle (1,2,3,4), (1,3) \rangle
gap > c4 := Subgroup(d8, [(1,2,3,4)]);;
gap> k4 := Subgroup( d8, [ (1,3), (2,4) ] );;
gap> SetName( c4, "c4" ); SetName( k4, "k4" );
gap> Ud8 := Subgroupoid( Gd8, [ [ k4, [-9] ], [ c4, [-8,-7] ] ] );;
gap> SetName( Ud8, "Ud8" );
gap> Display( Ud8 );
groupoid with 2 pieces:
< objects: [ -9 ]</pre>
    group: k4 = \langle (1,3), (2,4) \rangle \rangle
< objects: [ -8, -7 ]
    group: c4 = \langle (1,2,3,4) \rangle \rangle
gap> [ Parent( Ud8 ), IsWideSubgroupoid( Gd8, Ud8 ) ];
[ Gd8, true ]
gap> U2;
groupoid with 2 pieces:
[ Gf2, Gq8 ]
gap> genf2b := List( GeneratorsOfGroup(f2), g -> g^2 );
[ f1^2, f2^2 ]
gap> f2b := Subgroup( f2, genf2b );;
gap> SU2 := SubgroupoidByPieces( U2, [ [q8,[-17]], [f2b,[-22]] ] );
groupoid with 2 pieces:
1: single piece groupoid: < Group( [ f1^2, f2^2 ] ), [ -22 ] >
2: single piece groupoid: < q8, [ -17 ] >
gap> IsWideSubgroupoid( U2, SU2 );
gap> IsSubgroupoid( Gf2, Groupoid( f2b, [-22] ) );
true
```

## 4.3.5 FullTrivialSubgroupoid

```
    ▷ FullTrivialSubgroupoid(gpd) (attribute)
    ▷ DiscreteTrivialSubgroupoid(gpd) (attribute)
```

A *trivial subgroupoid* has trivial object groups, but need not be discrete. A single piece trivial groupoid is sometimes called a *tree groupoid*. (The term *identity subgroupoid* was used in versions up to 1.14.)

```
gap> FullTrivialSubgroupoid( Ud8 );
groupoid with 2 pieces:
```

```
1: single piece groupoid: < id(k4), [ -9 ] >
2: single piece groupoid: < id(c4), [ -8, -7 ] >
gap> DiscreteTrivialSubgroupoid( Ud8 );
groupoid with 3 pieces:
1: single piece groupoid: < id(k4), [ -9 ] >
2: single piece groupoid: < id(c4), [ -8 ] >
3: single piece groupoid: < id(c4), [ -7 ] >
```

## 4.3.6 DiscreteSubgroupoid

```
▷ DiscreteSubgroupoid(gpd, sgps, obs) (operation)
▷ MaximalDiscreteSubgroupoid(gpd) (attribute)
```

A subgroupoid is *discrete* if it is a union of groups. The MaximalDiscreteSubgroupoid of gpd is the union of all the single-object full subgroupoids of gpd.

```
gap> U3;
groupoid with 3 pieces:
[ Gs4, Gd8, Gc6 ]
gap> DiscreteSubgroupoid( U3, [ a4, a4, c4, k4 ], [-15,-11,-9,-7] );
groupoid with 4 pieces:
1: single piece groupoid: < a4, [ -15 ] >
2: single piece groupoid: < a4, [ -11 ] >
3: single piece groupoid: < c4, [ -9 ] >
4: single piece groupoid: < k4, [ -7 ] >
gap> MaximalDiscreteSubgroupoid( Vc3 );
groupoid with 3 pieces:
1: single piece groupoid: < c3, [ -14 ] >
2: single piece groupoid: < Group( [ (1,2,3) ] ), [ -13 ] >
3: single piece groupoid: < Group( [ (1,4,2) ] ), [ -12 ] >
```

## 4.3.7 SinglePieceSubgroupoidByGenerators

```
▷ SinglePieceSubgroupoidByGenerators(parent, gens) (operation)
```

A set of arrows generates a groupoid by taking all possible products and inverses. So far, the only implementation is for the case of loops generating a group at an object o together with a set of rays from o, where o is *not* the least object. A suitably large supergroupoid, which must be a direct product with a complete digraph, should be provided. This is the case needed for ConjugateGroupoid in section 4.5.2. Other cases will be added as time permits.

```
gap> a1 := Arrow( Ua4, (2,3,4), -15, -15 );
gap> a2 := Arrow( Ua4, (1,2,3,4), -15, -13 );
gap> a3 := Arrow( Ua4, (2,3), -15, -11 );
gap> SinglePieceSubgroupoidByGenerators( Ua4, [a1,a2,a3] );
```

```
single piece groupoid with rays: < Group( [ (2,3,4) ] ), [ -15, -13, -11 ],
[ (), (1,2,3,4), (2,3) ] >
```

## 4.4 Left, right and double cosets

## 4.4.1 RightCoset

```
▷ RightCoset(G, U, elt)
                                                                                       (operation)
▷ RightCosetRepresentatives(G, U)
                                                                                       (operation)

▷ RightCosets(G, U)
                                                                                       (operation)
▷ LeftCoset(G, U, elt)
                                                                                       (operation)
▷ LeftCosetRepresentatives(G, U)
                                                                                       (operation)
▷ LeftCosetRepresentativesFromObject(G, U, obj)
                                                                                       (operation)
▷ LeftCosets(G, U)
                                                                                       (operation)
▷ DoubleCoset(G, U, V, elt)
                                                                                       (operation)
▷ DoubleCosetRepresentatives(G, U, V)
                                                                                       (operation)
\triangleright DoubleCosets(G, U, V)
                                                                                       (operation)
```

If U is a subgroupoid of G, the *right cosets* Ug of U in G are the equivalence classes for the relation on the arrows of G where g1 is related to g2 if and only if g2 = u \* g1 for some arrow u of U. The right coset containing g is written Ug. These right cosets partition the costars of G and, in particular, the costar  $U1\_o$  of U at object o. So (unlike groups) U is itself a coset only when G has a single object.

The right coset representatives for U in G form a list containing one arrow for each coset where, in a particular piece of U, the group element chosen is the right coset representative of the group of U in the group of G.

Similarly, the *left cosets gU* refine the stars of G while *double cosets* are unions of left and right cosets. The operation LeftCosetRepresentativesFromObject(G, U, obj ) is used in Chapter G, and returns only those representatives which have tail at obj.

As with stars and homsets, these cosets are implemented with representation  ${\tt IsHomsetCosetsRep}$  and provided with an iterator. Note that, when U has more than one piece, cosets may have differing lengths.

In the example the representative for the right coset re2 is the tenth one in the printed list rcrd8, namely [(2,4):-7->-7].

Note that operations for double cosets are a recent addition, and may need further work.

```
gap> re2 := RightCoset( Gd8, Ud8, e2 );

<right coset of single piece groupoid: < c4, [ -8, -7 ] >

with representative [(1,3) : -8 -> -7]>

gap> Perform( re2, Display );

[(1,3) : -8 -> -7]

[(1,3) : -7 -> -7]

[(2,4) : -8 -> -7]

[(2,4) : -8 -> -7]

[(1,4)(2,3) : -8 -> -7]

[(1,4)(2,3) : -8 -> -7]

[(1,2)(3,4) : -8 -> -7]
```

```
[(1,2)(3,4): -7 \rightarrow -7]
gap> rcrd8 := RightCosetRepresentatives( Gd8, Ud8 );
[[(): -9 \rightarrow -9], [(1,4,3,2): -9 \rightarrow -9], [(): -9 \rightarrow -8],
  [(1,4,3,2): -9 \rightarrow -8], [(): -9 \rightarrow -7], [(1,4,3,2): -9 \rightarrow -7],
  [(): -8 -> -8], [(2,4): -8 -> -8], [(): -7 -> -7], [(2,4): -7 -> -7],
  [(): -8 \rightarrow -9], [(2,4): -8 \rightarrow -9]]
gap> le2 := LeftCoset( Gd8, Ud8, e2 );
<left coset of single piece groupoid: < c4, [ -8, -7 ] > with representative [
(1,3) : -8 -> -8]>
gap> Perform( le2, Display );
[(1,3): -8 \rightarrow -8]
[(1,3): -8 \rightarrow -7]
[(2,4) : -8 -> -8]
[(2,4) : -8 \rightarrow -7]
[(1,4)(2,3): -8 \rightarrow -8]
[(1,4)(2,3) : -8 \rightarrow -7]
[(1,2)(3,4) : -8 \rightarrow -8]
[(1,2)(3,4): -8 \rightarrow -7]
gap> lcrd8 := LeftCosetRepresentatives( Gd8, Ud8 );
[[():-9 \rightarrow -9], [(1,2,3,4):-9 \rightarrow -9], [():-8 \rightarrow -9],
  [(1,2,3,4) : -8 \rightarrow -9], [() : -7 \rightarrow -9], [(1,2,3,4) : -7 \rightarrow -9],
  [(): -8 -> -8], [(2,4): -8 -> -8], [(): -7 -> -7], [(2,4): -7 -> -7],
  [() : -9 \rightarrow -8], [(2,4) : -9 \rightarrow -8]]
gap> lcr7 := LeftCosetRepresentativesFromObject( Gd8, Ud8, -7 );
[[():-7->-9],[(1,2,3,4):-7->-9],[():-7->-7],
  [(2,4):-7->-7]
gap> de2 := DoubleCoset( Gd8, Ud8, Ud8, e2 );
<double coset of [ single piece groupoid: < c4, [ -8, -7 ] >,
  single piece groupoid: < c4, [-8, -7] > ] with representative [(1,3):
-8 -> -8]>
gap> Perform( de2, Display );
[(2,4) : -8 -> -8]
[(2,4): -8 \rightarrow -7]
[(2,4): -7 -> -8]
[(2,4): -7 \rightarrow -7]
[(1,3): -8 -> -8]
[(1,3): -8 -> -7]
[(1,3) : -7 -> -8]
[(1,3): -7 \rightarrow -7]
[(1,2)(3,4) : -8 \rightarrow -8]
[(1,2)(3,4): -8 \rightarrow -7]
[(1,2)(3,4): -7 \rightarrow -8]
[(1,2)(3,4): -7 \rightarrow -7]
[(1,4)(2,3) : -8 \rightarrow -8]
[(1,4)(2,3) : -8 \rightarrow -7]
[(1,4)(2,3) : -7 \rightarrow -8]
[(1,4)(2,3): -7 \rightarrow -7]
gap> dcrd8 := DoubleCosetRepresentatives( Gd8, Ud8, Ud8 );
[(): -9 -> -9], [(1,4,3,2): -9 -> -9], [(): -9 -> -8], [(): -8 -> -9],
  [(): -8 \rightarrow -8], [(2,4): -8 \rightarrow -8]]
```

## 4.5 Conjugation

## 4.5.1 \^

▷ \^(e1, e)

Conjugation by an arrow  $e = (c : p \to q)$  is the groupoid inner automorphism (see section 5.6) defined as follows. There are two cases to consider. In the case  $p \neq q$ ,

- objects p,q are interchanged, and the remaining objects are fixed;
- loops at p,q:  $(b:p\to p)\mapsto (b^c:q\to q)$  and  $(b:q\to q)\mapsto (b^{c^{-1}}:p\to p)$ ;
- arrows between p and q:  $(b:p\to q)\mapsto (c^{-1}bc^{-1}:q\to p)$  and  $(b:q\to p)\mapsto (cbc:p\to q)$ ;
- costars at  $p,q:(b:r\to p)\mapsto (bc:r\to q)$  and  $(b:r\to q)\mapsto (bc^{-1}:r\to p)$ ;
- stars at  $p,q:(b:p\to r)\mapsto (c^{-1}b:\to q)$  and  $(b:q\to r)\mapsto (cb:p\to r)$ ;
- the remaining arrows are unchanged.

In the case p = q,

- all the objects are fixed;
- loops at p are conjugated by c, so  $(b: p \to p) \mapsto (b^c: p \to p)$ ;
- the rest of the costar and star at p are permuted,

$$(b:r\to p)\mapsto (bc:r\to p)$$
 and  $(b:p\to r)\mapsto (c^{-1}b:p\to r);$ 

• the remaining arrows are unchanged.

The details of this construction may be found in section 3.2 of [AW10].

```
_{-} Example
gap> x := Arrow(Gd8, (1,3), -9, -9);;
gap> y := Arrow( Gd8, (1,2,3,4), -8, -9 );;
gap> z := Arrow(Gd8, (1,2)(3,4), -9, -7);;
gap> w := Arrow( Gd8, (1,2,3,4), -7, -8 );;
gap> ## conjugation with elements x, y, and z in Gd8:
gap> x^y;
[(2,4) : -8 \rightarrow -8]
gap> x^z;
[(2,4): -7 \rightarrow -7]
gap> y^x;
[(): -8 -> -9]
gap> y^z;
[(2,4): -8 \rightarrow -7]
gap> z^x;
[(1,4,3,2) : -9 \rightarrow -7]
gap> z^y;
[(2,4): -8 \rightarrow -7]
gap> w^z
[(1,3): -9 -> -8]
```

## 4.5.2 ConjugateGroupoid

```
▷ ConjugateGroupoid(gpd, e)
```

(operation)

When H is a subgroupoid of a groupoid G and a is an arrow of G, then the conjugate of H by a is the subgroupoid generated by the conjugates of the generators of H.

```
gap> Kd8 := SubgroupoidWithRays( Gs4, d8, [(),(1,2),(2,3),(3,4),(1,4)] );
single piece groupoid with rays: < d8, [ -15 .. -11 ],
[ (), (1,2), (2,3), (3,4), (1,4) ] >
gap> u := Arrow( Gs4, (1,2,3), -15, -14 );
[(1,2,3) : -15 -> -14]
gap> ConjugateGroupoid( Kd8, u );
single piece groupoid with rays: < Group( [ (1,3,2,4), (1,2) ] ),
[ -15, -14, -13, -12, -11 ], [ (), (1,2), (), (2,4,3), (1,4)(2,3) ] >
```

## 4.6 Groupoids formed using isomorphisms

Here we describe an alternative way of constructing a connected groupoid. This section has been introduced in version 1.66 of the package, and so should be considered very experimental, and liable to change.

Object groups in a connected groupoid are isomorphic, so we may use a collection of isomorphisms to form a groupoid. Let  $G_1, G_2, \ldots, G_n$  be isomorphic groups and, for  $2 \le i \le n$ , let  $\mu_i : G_1 \to G_i$  be isomorphisms. Then  $\mu_{ij} = \mu_i^{-1} * \mu_j$  is an isomorphism from  $G_i$  to  $G_j$ . If we take  $\{u_1, \ldots, u_n\}$  to be our set of objects, with  $G_i$  the object group at  $u_i$ , we may consider the arrows in the groupoid to have the form  $[[g_i, g_j] : u_i \to u_j]$  where  $g_i \in G_i$  and  $g_j = \mu_{ij}(g_i) \in G_j$ . The product of  $[[g_i, g_j] : u_i \to u_j]$  and  $[[g'_j, g_k] : u_j \to u_k]$  is  $[[\mu_{ij}^{-1}(g_jg'_j), \mu_{jk}(g_jg'_j)] : u_i \to u_k]$ .

## 4.6.1 GroupoidByIsomorphisms

```
▷ GroupoidByIsomorphisms(gp, obs, isos) (operation)
▷ IsGroupoidByIsomorphisms(gpd) (property)
```

The operation GroupoidByIsomorphisms takes a group G1 as root group; a set of n objects; and a set of n isomorphisms from the root group, where the first isomorphism should be the identity mapping on G1. The output is a single piece groupoid of type IsGroupoidByIsomorphisms. Its rays have the form [One(G1), One(Gi)] where Gi is the image of the i-th isomorphism.

In the example we first take three permutation groups isomorphic to the symmetric group  $S_3$ . There follows an isomorphic groupoid whose object groups are a permutation group; a pc-group and an fp-group.

```
gap> s3a := Group( (1,2), (2,3) );;
gap> s3b := Group( (4,6,8)(5,7,9), (4,9)(5,8)(6,7) );;
gap> s3c := Group( (4,6,8)(5,7,9), (5,9)(6,8) );;
gap> SetName( s3a, "s3a" );
```

```
gap> ida := IdentityMapping( s3a );;
gap> isoab := IsomorphismGroups( s3a, s3b );;
gap> isoac := IsomorphismGroups( s3a, s3c );;
gap> isos1 := [ ida, isoab, isoac ];;
gap> G1 := GroupoidByIsomorphisms( s3a, [-3,-2,-1], isos1 );;
gap> gens1 := GeneratorsOfGroupoid( G1 );
[[[(1,2), (1,2)] : -3 \rightarrow -3], [[(2,3), (2,3)] : -3 \rightarrow -3],
  [[(), ()] : -3 \rightarrow -2], [[(), ()] : -3 \rightarrow -1]]
gap> x1 := ImageElm( isos1[2], (1,2) );;
gap> a1 := Arrow( G1, [ (1,2), x1 ], -3, -2 );
[[(1,2), (4,5)(6,9)(7,8)] : -3 \rightarrow -2]
gap> a1^-1;
[[ (4,5)(6,9)(7,8), (1,2) ] : -2 \rightarrow -3]
gap> y1 := ImageElm( isos1[2], (2,3) );;
gap> z1 := ImageElm( isos1[3], (2,3) );;
gap> b1 := Arrow( G1, [ y1, z1 ], -2, -1 );
[[ (4,9)(5,8)(6,7), (5,9)(6,8) ] : -2 \rightarrow -1]
gap> c1 := a1*b1;
[[(1,3,2), (4,8,6)(5,9,7)]: -3 \rightarrow -1]
gap> isopc := IsomorphismPcGroup( s3a );;
gap> s3p := Image( isopc );;
gap> f2 := FreeGroup( 2 );;
gap> s3f := f2/[f2.1^3, f2.2^2, (f2.1*f2.2)^2];;
gap> isofp := GroupHomomorphismByImages(s3a,s3f,[(1,2,3),(2,3)],[s3f.1,s3f.2]);;
gap> isos2 := [ ida, isopc, isofp ];;
gap> G2 := GroupoidByIsomorphisms( s3a, [-7,-6,-5], isos2 );;
gap> gens2 := GeneratorsOfGroupoid( G2 );
[[[(1,2), (1,2)] : -7 \rightarrow -7], [[(2,3), (2,3)] : -7 \rightarrow -7],
  [[(), <identity> of ...]: -7 -> -6], [[(), <identity ...>]: -7 -> -5]
gap> x2 := ImageElm( isos2[2], (1,2) );;
gap > a2 := Arrow(G2, [(1,2), x2], -7, -6);
[[(1,2), f1*f2] : -7 -> -6]
gap> a2^-1;
[[f1*f2, (1,2)] : -6 \rightarrow -7]
gap> y2 := ImageElm( isos2[2], (2,3) );;
gap> z2 := ImageElm( isos2[3], (2,3) );;
gap> b2 := Arrow( G2, [ y2, z2 ], -6, -5 );
[[ f1, f2^-1 ] : -6 -> -5]
gap> c2 := a2*b2;
[[(1,3,2), f1^2]: -7 \rightarrow -5]
```

## 4.7 Groupoids whose objects form a monoid

Let M be a monoid with G its maximal subgroup. We may form a groupoid with the elements of M as its objects and with arrows  $t \to t * g$  for all  $t \in M$  and  $g \in G$ .

This construction is used in the XMod package to construct the group-groupoid which corresponds to a crossed module or cat2-group.

## 4.7.1 SinglePieceGroupoidWithRays

```
▷ SinglePieceGroupoidWithRays(gp, obs, rays)
```

(operation)

When M is a group, G = M and the groupoid so constructed is a single piece which represents the regular representation of G. The ray from 1 to g is just g since 1 \* g = g.

```
Example
gap> d8 := Group( (1,2,3,4), (1,3) );;
gap> ed8 := Elements( d8 );;
gap> rd8 := SinglePieceGroupoidWithRays( d8, ed8, ed8 );
single piece groupoid with rays: < Group([(1,2,3,4), (1,3)]),
[(1), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2), (1,4)(2,3)
], [(), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2),
  (1,4)(2,3) ] >
gap> Display( Homset( rd8, (2,4), (1,3) ) );
[(1,3)(2,4):(2,4)\rightarrow(1,3)]
[(1,3):(2,4)\to(1,3)]
[():(2,4)\rightarrow(1,3)]
[(2,4) : (2,4) \rightarrow (1,3)]
[(1,4,3,2) : (2,4) \rightarrow (1,3)]
[(1,4)(2,3):(2,4)\rightarrow(1,3)]
[(1,2,3,4) : (2,4) \rightarrow (1,3)]
[(1,2)(3,4):(2,4)\rightarrow(1,3)]
```

## 4.7.2 RightActionGroupoid

```
▷ RightActionGroupoid(mon) (operation)
▷ IsGroupoidWithMonoidObjects(gpd) (property)
```

When M is a monoid, rather than a group, this construction produces several components. One of these has as objects the elements of the group G.

When M is a group, this operation gives the same result as SinglePieceGroupoidWithRays.

As a simple example we take a monoid M generated by two transformations of degree 4. The groupoid has 8 components, of which 3 have a single object and group  $C_2$ , while 5 have two objects and trivial group.

```
gap> t := Transformation( [1,1,2,3] );; u := Transformation( [1,2,4,3] );;
gap> M := Monoid( t, u );;
gap> rag := RightActionGroupoid( M );
groupoid with 8 pieces:
1: single piece groupoid with rays: < Group(
  [ IdentityTransformation, Transformation( [ 1, 2, 4, 3 ] ) ] ),
  [ Transformation( [ 1, 1, 1, 1 ] ) ], [ IdentityTransformation ] >
2: single piece groupoid with rays: < Group(
  [ IdentityTransformation, Transformation( [ 1, 2, 4, 3 ] ) ] ),
  [ Transformation( [ 1, 1, 1, 2 ] ) ], [ IdentityTransformation ] >
3: single piece groupoid with rays: < Group( [ IdentityTransformation ] ),</pre>
```

```
[ Transformation( [ 1, 1, 1, 3 ] ), Transformation( [ 1, 1, 1 ] ) ],
[ IdentityTransformation, Transformation([1, 2, 4, 3])] >
4: single piece groupoid with rays: < Group(
[ IdentityTransformation, Transformation([1, 2, 4, 3])]),
[ Transformation([1, 1, 2, 1])], [ IdentityTransformation] >
5: single piece groupoid with rays: < Group( [ IdentityTransformation ] ),
[ Transformation([1, 1, 2, 3]), Transformation([1, 1, 2])],
[ IdentityTransformation, Transformation([1, 2, 4, 3])] >
6: single piece groupoid with rays: < Group( [ IdentityTransformation ] ),
[ Transformation([1, 1, 3, 1]), Transformation([1, 1, 4, 1])],
[ IdentityTransformation, Transformation([1, 2, 4, 3])] >
7: single piece groupoid with rays: < Group( [ IdentityTransformation ] ),
[ Transformation([1, 1, 3, 2]), Transformation([1, 1, 4, 2])],
[ IdentityTransformation, Transformation([1, 2, 4, 3])] >
8: single piece groupoid with rays: < Group( [ IdentityTransformation ] ),
[ IdentityTransformation, Transformation( [ 1, 2, 4, 3 ] ) ],
[ IdentityTransformation, Transformation([1, 2, 4, 3])] >
gap> IsGroupoidWithMonoidObjects( rag );
gap> orag := ObjectList( rag );;
gap> hs := Homset( rag, orag[3], orag[4] );;
gap> Display( hs );
<homset Transformation([ 1, 1, 1, 3 ] ) -> Transformation([ 1, 1, 1 ] )
  with elements:
[Transformation([1, 2, 4, 3]): Transformation([1, 1, 1, 3]) ->
Transformation([1, 1, 1])]
```

# Chapter 5

# **Homomorphisms of Groupoids**

A homomorphism m from a groupoid G to a groupoid H consists of a map from the objects of G to those of H together with a map from the elements of G to those of H which is compatible with tail and head and which preserves multiplication:

$$m(g1:o1 \to o2)*m(g2:o2 \to o3) = m(g1*g2:o1 \to o3).$$

Note that when a homomorphism is not injective on objects, the image of the source need not be a subgroupoid of the range. A simple example of this is given by a homomorphism from the two-object, four-element groupoid with trivial group to the free group  $\langle a \rangle$  on one generator, when the image is  $[1, a^n, a^{-n}]$  for some n > 0.

A variety of homomorphism operations are available.

- The basic construction is a homomorphism  $\phi: G \to H$  from a connected groupoid G to a connected groupoid H, constructed using GroupoidHomomorphismFromSinglePiece, (see 5.1).
- Since more than one connected groupoid may be mapped *to* the same range, we then have the operation GroupoidHomomorphismToSinglePiece, (see 5.4).
- The third case arises when the range is a union of connected groupoids, in which case HomomorphismByUnion is called, (see 5.5).
- Fourthly, there are is an additional operation for the case where the source is homogeneous and discrete, GroupoidHomomorphismFromHomogeneousDiscrete, (see 5.4.2).
- Finally, there are special operations for inclusion mappings, restricted mappings, and groupoid automorphisms (see 5.6).

### 5.1 Homomorphisms from a connected groupoid

### 5.1.1 GroupoidHomomorphismFromSinglePiece

▷ GroupoidHomomorphismFromSinglePiece(src, rng, gens, images) (operation)
▷ GroupoidHomomorphism(args) (function)
▷ MappingToSinglePieceData(map) (attribute)

The simplest groupoid homomorphism is a mapping  $\phi: G \to H$  from a connected groupoid G to a connected groupoid H. There are two equivalent sets of input data which may be used. Both require the Source G and the Range H. The first then requires:

- the set of generating arrows GeneratorsOfGroupoid(G);
- a list of image arrows in H.

This data is stored in the attribute MappingGeneratorsImages.

The alternative input data consists of:

- a homomorphism rhom from the root group of G to the group at the image object in H;
- a list imobs of the images of the objects of G;
- a list imrays of the elements in the images of the rays of G, so that the image  $\phi(r_i: o_1 \to o_i)$  of the *i*-th ray is (imrays[i]:imobs[1] $\to$ imobs[i]).

This data is stored in the attribute MappingToSinglePieceData.

So an alternative way to construct a homomorphism of groupoids is to make a call of the form GroupoidHomomorphism(src,rng,rhom,imobs,imrays).

In the following example the same homomorphism is constructed using both methods.

```
Example -
gap> gen1 := GeneratorsOfGroupoid( Gq8 );
[[x : -28 \rightarrow -28], [y : -28 \rightarrow -28], [y2 : -28 \rightarrow -28],
  [<identity> of ... : -28 -> -27] ]
gap> gen2 := GeneratorsOfGroupoid( Hd8b );
[(1,2,3,4): -14 \rightarrow -14], [(1,3): -14 \rightarrow -14], [(1,2,3): -14 \rightarrow -13],
  [(1,2,4): -14 \rightarrow -12]]
gap> images := [ gen2[1]^2, gen2[1]*gen2[2], IdentityArrow(Hd8b,-14), gen2[4] ];
[[(1,3)(2,4):-14 \rightarrow -14], [(1,2)(3,4):-14 \rightarrow -14], [():-14 \rightarrow -14],
  [(1,2,4): -14 \rightarrow -12]]
gap> mor1 := GroupoidHomomorphism( Gq8, Hd8b, gen1, images );
groupoid homomorphism : Gq8 -> Hd8b
[ [ [x : -28 \rightarrow -28], [y : -28 \rightarrow -28], [y2 : -28 \rightarrow -28],
       [<identity> of ... : -28 -> -27],
  [[(1,3)(2,4): -14 \rightarrow -14], [(1,2)(3,4): -14 \rightarrow -14], [(): -14 \rightarrow -14],
       [(1,2,4) : -14 \rightarrow -12]]
gap> genq8 := GeneratorsOfGroup( q8 );;
gap> imh := [(1,3)(2,4), (1,2)(3,4), ()];;
gap> h := GroupHomomorphismByImages( q8, d8, genq8, imh );
[x, y, y2] \rightarrow [(1,3)(2,4), (1,2)(3,4), ()]
gap> mor2 := GroupoidHomomorphism( Gq8, Hd8b, h, [-14,-12], [(),(1,2,4)] );;
gap> mor1=mor2;
true
gap> e := Arrow( Gq8, Product(genq8), -27, -28 );
[x*y*y2 : -27 -> -28]
gap> ImageElm( mor2, e );
[(2,4,3): -12 \rightarrow -14]
```

## 5.2 Properties and attributes of groupoid homomorphisms

### 5.2.1 Properties of a groupoid homomorphism

The properties listed in subsection 3.3 for homomorphisms of magmas with objects also apply to groupoid homomorphisms.

```
_{-} Example
gap> [ IsInjectiveOnObjects( mor2 ), IsSurjectiveOnObjects( mor2 ) ];
[ true, false ]
gap> [ IsInjective( mor2 ), IsSurjective( mor2 ) ];
[false, false]
gap> ad8 := GroupHomomorphismByImages( d8, d8,
                 [(1,2,3,4), (1,3)], [(1,4,3,2), (2,4)]);;
gap> md8 := GroupoidHomomorphism( Gd8, Gd8, ad8, [-7,-9,-8], [(),(1,3),(2,4)] );
groupoid homomorphism : Gd8 -> Gd8
[[(1,2,3,4):-9 \rightarrow -9],[(1,3):-9 \rightarrow -9],[():-9 \rightarrow -8],
      [() : -9 \rightarrow -7]],
  [[(1,4,3,2) : -7 \rightarrow -7], [(2,4) : -7 \rightarrow -7], [(1,3) : -7 \rightarrow -9],
      [(2,4): -7 \rightarrow -8]]
gap> IsBijectiveOnObjects( md8 );
true
gap> [ IsInjective( md8 ), IsSurjective( md8 ) ];
[ true, true ]
gap> [ IsEndomorphismWithObjects( md8 ), IsAutomorphismWithObjects( md8 ) ];
```

### 5.2.2 Attributes of a groupoid homomorphism

The attributes of a groupoid homomorphism mor from a single piece groupoid cover both forms of construction defined above.

- S = Source(mor) is the source groupoid of the homomorphism;
- R = Range(mor) is the range groupoid of the homomorphism;
- RootGroupHomomorphism(mor) is the group homomorphism from the root group of S to the group at the image object in R of the root object in S;
- ImagesOfObjects (mor) is the list of objects in R which are the images of the objects in S;
- ImageElementsOfRays(mor) is the list of group elements in those arrows in R which are the images of the rays in S;
- MappingGeneratorsImages (mor) is the two element list containing the list of generators in S and the list of their images in R;
- MappingToSinglePieceData(mor) is a list with three elements: the root group homomorphism; the images of the objects; and the images of the rays.

For other types of homomorphism the attributes are very similar.

The operation ObjectGroupHomomorphism, though an operation, is included in this section for convenience.

### 5.2.3 RootGroupHomomorphism

```
▷ RootGroupHomomorphism(hom)
```

(attribute)

This is the group homomorphism from the root group of the source groupoid to the group at the image object in the range groupoid of the root object in the source.

### 5.2.4 ImagesOfObjects

```
▷ ImagesOfObjects(hom)
```

(attribute)

This is the list of objects in the range groupoid which are the images of the objects in the source.

### 5.2.5 ImageElementsOfRays

```
▷ ImageElementsOfRays(hom)
```

(attribute)

This is the list of group elements in those arrows in the range groupoid which are the images of the rays in the source.

```
gap> RootGroupHomomorphism( mor2 );
[ x, y, y2 ] -> [ (1,3)(2,4), (1,2)(3,4), () ]
gap> ImagesOfObjects( mor2 );
[ -14, -12 ]
gap> ImageElementsOfRays( mor2 );
[ (), (1,2,4) ]
```

### 5.2.6 ObjectGroupHomomorphism

```
▷ ObjectGroupHomomorphism(gpdhom, obj)
```

(operation)

This operations gives the group homomorphism from an object group of the source to the object group at the image object in the range.

```
gap> ObjectGroupHomomorphism( mor1, -27 );
[ x, y, y2 ] -> [ (1,4)(2,3), (1,3)(2,4), () ]
```

# 5.3 Special types of groupoid homomorphism

In this section we mention inclusion mappings of subgroupoids; and mappings restricted to a source subgroupoid. We also discuss various types of isomorphism: to a different set of objects; to a permutation groupoid; to a pc-groupoid.

### 5.3.1 InclusionMappingGroupoids

```
▷ InclusionMappingGroupoids(gpd, sgpd)
```

(operation)

The operation InclusionMappingGroupoids(gpd,sgpd) returns the inclusion homomorphism from the subgroupoid sgpd to gpd.

### 5.3.2 RestrictedMappingGroupoids

```
▷ RestrictedMappingGroupoids(mor, sgpd) (operation)
▷ ParentMappingGroupoids(mor) (attribute)
```

The operation RestrictedMappingGroupoids (mor, sgpd) returns the restriction of the homomorphism mor to the subgroupoid sgpd of its source. The range is usually set to the ImagesSource of the restriction. The restriction is assigned the attribute ParentMappingGroupoids with value mor (or that of mor is one exists). For another example see section 5.7.

```
gap> max := MaximalDiscreteSubgroupoid( Hd8b );;
gap> res := RestrictedMappingGroupoids( inc, max );
groupoid homomorphism from several pieces :
groupoid homomorphism :
[ [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14] ],
        [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14] ] ]
groupoid homomorphism :
[ [ [(1,4,2,3) : -13 -> -13], [(1,2) : -13 -> -13] ],
        [ [(1,4,2,3) : -13 -> -13], [(1,2) : -13 -> -13] ] ]
groupoid homomorphism :
[ [ [(1,2,4,3) : -12 -> -12], [(2,3) : -12 -> -12] ],
        [ [(1,2,4,3) : -12 -> -12], [(2,3) : -12 -> -12] ] ]
```

### 5.3.3 IsomorphismNewObjects (for groupoids)

```
▷ IsomorphismNewObjects(src, objlist)
```

(operation)

The operation IsomorphismNewObjects(gpd,obs) returns the isomorphism from a groupoid gpd to a groupoid with the same object group and ray elements but with a different set obs of objects.

```
gap> iso1 := IsomorphismNewObjects( Hs4, [-30,-20,-10] );
```

### 5.3.4 IsomorphismStandardGroupoid

 ${\tt \vartriangleright IsomorphismStandardGroupoid}(\mathit{gpd},\ \mathit{obs})$ 

(operation)

The operation IsomorphismStandardGroupoid(gpd,obs) returns the isomorphism from a groupoid with rays to the groupoid of type IsDirectProductWithCompleteDigraphDomain on the given set obs of objects.

This operation may also be used to provide a standard form for groupoids of type IsGroupoidByIsomorphisms as described in subsection GroupoidByIsomorphisms (4.6.1).

### 5.3.5 IsomorphismPermGroupoid

The operation IsomorphismPermGroupoid(gpd) returns an isomorphism from a groupoid gpd to a groupoid with the same objects but with an isomorphic permutation group. Similarly, IsomorphismPcGroupoid(gpd) changes the group into a pc-group (if appropriate).

```
gap> N2 := NormalSubgroups( q8 )[2];;
gap> Hq8 := SubgroupoidWithRays( Gq8, N2, [ One(q8), q8.1 ] );
single piece groupoid with rays: < N2, [ -28, -27 ], [ <identity> of ..., x
] >
gap> isoHq8 := IsomorphismPermGroupoid( Hq8 );
groupoid homomorphism :
[ [ y : -28 -> -28], [y2 : -28 -> -28], [x : -28 -> -27] ],
        [ [(1,3,4,7)(2,5,6,8) : -28 -> -28], [(1,4)(2,6)(3,7)(5,8) : -28 -> -28],
        [(1,2,4,6)(3,8,7,5) : -28 -> -27] ] ]
```

## 5.4 Homomorphisms to a connected groupoid

### **5.4.1** HomomorphismToSinglePiece (for groupoids)

```
→ HomomorphismToSinglePiece(src, rng, piecehoms) (operation)
```

When *G* is made up of two or more pieces, all of which get mapped to a connected groupoid, we have a *homomorphism to a single piece*. The third input parameter in this case is a list of the individual homomorphisms *from* the single pieces (in the correct order!). See section 3.1 for the corresponding operation on homomorphisms of magmas with objects.

In the following example the source V3 of homV3 has three pieces, and one of the component homomorphisms is an IdentityMapping.

```
_{-} Example
gap> gend12 := [ (15,16,17,18,19,20), (15,20)(16,19)(17,18) ];;
gap> d12 := Group( gend12 );;
gap> Gd12 := Groupoid( d12, [-37,-36,-35,-34] );;
gap> SetName( d12, "d12" );
gap> SetName( Gd12, "Gd12" );
gap> s3 := Subgroup( d12, [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] );;
gap> Gs3 := SubgroupoidByPieces( Gd12, [ [ s3, [-36,-35,-34] ] ] );;
gap> SetName( s3, "s3" );
gap> SetName( Gs3, "Gs3" );
gap> gend8 := GeneratorsOfGroup( d8 );;
gap> imhd8 := [ ( ), (15,20)(16,19)(17,18) ];;
gap> hd8 := GroupHomomorphismByImages( d8, s3, gend8, imhd8 );;
gap> homd8 := GroupoidHomomorphism( Gd8, Gs3, hd8 );
groupoid homomorphism : Gd8 -> Gs3
[[(1,2,3,4):-9 \rightarrow -9],[(1,3):-9 \rightarrow -9],[():-9 \rightarrow -8],
```

```
[(): -9 -> -7]],
  [[(): -36 \rightarrow -36], [(15,20)(16,19)(17,18): -36 \rightarrow -36],
      [(): -36 \rightarrow -35], [(): -36 \rightarrow -34]]
gap> hc6 := GroupHomomorphismByImages( c6, s3,
             [(5,6,7)(8,9)], [(15,16)(17,20)(18,19)]);;
gap> Fs3 := SubgroupoidByObjects( Gs3, [ -35 ] );;
gap> SetName( Fs3, "Fs3" );
gap> homc6 := GroupoidHomomorphism( Gc6, Fs3, hc6 );;
gap> incFs3 := InclusionMappingGroupoids( Gs3, Fs3 );;
gap> ihomc6 := homc6 * incFs3;
groupoid homomorphism : Gc6 -> Gs3
[[(5,6,7)(8,9): -6 \rightarrow -6]], [(15,16)(17,20)(18,19): -35 \rightarrow -35]]
gap> idGs3 := IdentityMapping( Gs3 );;
gap> V3 := ReplaceOnePieceInUnion( U3, 1, Gs3 );
groupoid with 3 pieces:
[ Gs3, Gd8, Gc6 ]
gap> homs3 := [ idGs3, homd8, ihomc6 ];;
gap> homV3 := HomomorphismToSinglePiece( V3, Gs3, homs3 );;
gap> Display( homV3 );
homomorphism to single piece groupoid with pieces:
(1) : groupoid mapping: [ Gs3 ] -> [ Gs3 ]
root homomorphism: [ [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],
  [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] ]
images of objects: [-36, -35, -34]
   images of rays: [(): -36 -> -36], (): -36 -> -35], (): -36 -> -34]
(2): groupoid mapping: [ Gd8 ] -> [ Gs3 ]
root homomorphism: [ [ (1,2,3,4), (1,3) ], [ (), (15,20)(16,19)(17,18) ] ]
images of objects: [ -36, -35, -34 ]
   images of rays: [(): -36 -> -36], (): -36 -> -35], (): -36 -> -34]
(3): groupoid mapping: [ Gc6 ] -> [ Gs3 ]
root homomorphism: [ [ (5,6,7)(8,9) ], [ (15,16)(17,20)(18,19) ] ]
images of objects: [ -35 ]
   images of rays: [ [() : -35 -> -35] ]
```

#### 5.4.2 GroupoidHomomorphismFromHomogeneousDiscrete

□ GroupoidHomomorphismFromHomogeneousDiscrete(src, rng, homs, oims) (operation)

This operation requires the source and range; a list of homomorphisms from object group to object group; and a list of the image objects.

```
IdentityMapping( d8 )
IdentityMapping( d8 )
IdentityMapping( d8 )
```

# 5.5 Homomorphisms to more than one piece

### **5.5.1** HomomorphismByUnion (for groupoids)

(operation)

As in section 3.3, when the range H has more than one connected component, a homomorphism is a union of homomorphisms, one for each piece in the range.

```
_{-} Example
gap> isoq8 := IsomorphismNewObjects( Gq8, [-38,-37] );
groupoid homomorphism :
[ [ [x : -28 -> -28], [y : -28 -> -28], [y2 : -28 -> -28],
      [<identity> of ... : -28 -> -27]],
  [ [x : -38 \rightarrow -38], [y : -38 \rightarrow -38], [y2 : -38 \rightarrow -38],
      [<identity> of ... : -38 -> -37] ] ]
gap> Gq8b := Range( isoq8 );;
gap> SetName( Gq8b, "Gq8b" );
gap> V4 := UnionOfPieces( [ V3, Gq8 ] );
groupoid with 4 pieces:
[ Gs3, Gq8, Gd8, Gc6 ]
gap> SetName( V4, "V4" );
gap> Vs3q8b := UnionOfPieces( [ Gs3, Gq8b ] );;
gap> SetName( Vs3q8b, "Vs3q8b" );
gap> hom4 := HomomorphismByUnion( V4, Vs3q8b, [ homV3, isoq8 ] );;
gap> Display( hom4 );
magma homomorphism: V4 -> Vs3q8b with pieces :
[Pcgs([x, y, y2]) \rightarrow [x, y, y2], [-38, -37],
  [ <identity> of ..., <identity> of ...] ]
(1) : groupoid mapping: [ Gs3 ] -> [ Gs3 ]
root homomorphism: [ [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],
  [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] ]
images of objects: [ -36, -35, -34 ]
   images of rays: [(): -36 -> -36], (): -36 -> -35], (): -36 -> -34]
(2) : groupoid mapping: [ Gd8 ] -> [ Gs3 ]
root homomorphism: [ [ (1,2,3,4), (1,3) ], [ (), (15,20)(16,19)(17,18) ] ]
images of objects: [ -36, -35, -34 ]
   images of rays: [ [() : -36 -> -36], [() : -36 -> -35], [() : -36 -> -34] ]
(3) : groupoid mapping: [ Gc6 ] -> [ Gs3 ]
root homomorphism: [ [ (5,6,7)(8,9) ], [ (15,16)(17,20)(18,19) ] ]
images of objects: [ -35 ]
   images of rays: [ [() : -35 -> -35] ]
```

### 5.5.2 IsomorphismGroupoids

```
▷ IsomorphismGroupoids(A, B)
```

(operation)

When A, B are two single piece groupoids, they are isomorphic provided they have the same number of objects and the root groups are isomorphic.

When  $A = [A_1, ..., A_n]$ ,  $B = [B_1, ..., B_n]$  are both unions of connected groupoids, they are isomorphic if there is a permutation  $\pi$  of [1, ..., n] such that  $A_i$  is isomorphic to  $B_{\pi(i)}$  for all i.

```
_ Example
gap> s3a := Group( (1,2,3), (2,3) );;
gap> s3b := Group( (4,6,8)(5,7,9), (4,9)(5,8)(6,7));;
gap> s3c := Group( (4,6,8)(5,7,9), (5,9)(6,8));;
gap> Ga := SinglePieceGroupoid( s3a, [-23,-22,-21] );;
gap> SetName( Ga, "Ga" );
gap> Gb := SinglePieceGroupoid( s3b, [-6,-5,-4] );;
gap> SetName( Gb, "Gb" );
gap> Gc := SinglePieceGroupoid( s3c, [-9,-8,-7] );;
gap> SetName( Gc, "Gc" );
gap> c6a := Group( (1,2,3,4,5,6) );;
gap > c6b := Group((7,8)(9,10,11));;
gap > c6c := Group((12,13)(14,15)(16,17,18)(19,20,21)(22,23,24,25,26,27));;
gap> Ha := SinglePieceGroupoid( c6a, [-3,-2,-1] );;
gap> SetName( Ha, "Ha" );
gap> Hb := SinglePieceGroupoid( c6b, [-16,-15,-14] );;
gap> SetName( Hb, "Hb" );
gap> Hc := SinglePieceGroupoid( c6c, [-19,-18,-17] );;
gap> SetName( Hc, "Hc" );
gap> IsomorphismGroupoids( Ga, Gb );
groupoid homomorphism : Ga -> Gb
[ [ [(1,2,3) : -23 \rightarrow -23], [(2,3) : -23 \rightarrow -23], [() : -23 \rightarrow -22],
      [(): -23 \rightarrow -21]],
  [ [(4,6,8)(5,7,9) : -6 \rightarrow -6], [(4,9)(5,8)(6,7) : -6 \rightarrow -6],
      [(): -6 \rightarrow -5], [(): -6 \rightarrow -4]]
gap> IsomorphismGroupoids( Ga, Ha );
gap> A := UnionOfPieces( [ Ha, Gb, Gc, Hb ] );
groupoid with 4 pieces:
[ Hb, Gc, Gb, Ha ]
gap> B := UnionOfPieces( [ Gc, Hb, Hc, Ga ] );
groupoid with 4 pieces:
[ Ga, Hc, Hb, Gc ]
gap> iso := IsomorphismGroupoids( A, B );;
gap> Print( List( PiecesOfMapping(iso), p -> [Source(p),Range(p)] ) );
[ [ Gc, Ga ], [ Hb, Hc ], [ Ha, Hb ], [ Gb, Gc ] ]
```

# 5.6 Automorphisms of groupoids

In this sections we consider automorphisms of single piece groupoids, then homogeneous discrete groupoids, and then homogeneous groupoids.

### 5.6.1 GroupoidAutomorphismByObjectPerm

```
    ▷ GroupoidAutomorphismByObjectPerm(gpd, imobs)
    ▷ GroupoidAutomorphismByGroupAuto(gpd, gpiso)
    ▷ GroupoidAutomorphismByRayShifts(gpd, imrays)
    ▷ GroupoidInnerAutomorphism(gpd, arrow)
```

We first describe automorphisms of a groupoid G where G is the direct product of a group g and a complete digraph. The automorphism group is generated by three types of automorphism:

- given a permutation  $\pi$  of the n objects, we define  $\pi: G \to G, \ (o_i, g, o_j) \mapsto (o_{\pi i}, g, o_{\pi j});$
- given an automorphism  $\alpha$  of the root group g, we define  $\alpha: G \to G$ ,  $(o_i, g, o_j) \mapsto (o_i, \alpha g, o_j)$ ;
- given  $L = [1, g_2, g_3, \dots, g_n] \in g^n$  we define  $\theta_L : G \to G$ ,  $(o_i, g, o_j) \mapsto (o_i, g_i^{-1} g g_j, o_j)$  so that, in particular, for all j the rays  $(r_i : o_1 \to o_j)$  are shifted by  $g_j$ : they map to  $(r_i g_j : o_1 \to o_j)$ ;
- given  $g \in G$ , the *inner automorphism* of G by g is the mapping  $h \mapsto h^g$  wehere conjugation of arrows is defined in section 4.5.

```
\_ Example _-
gap> a4 := Subgroup( s4, [(1,2,3),(2,3,4)] );;
gap> SetName( a4, "a4" );
gap> gensa4 := GeneratorsOfGroup( a4 );;
gap> Ga4 := SubgroupoidByPieces( Gs4, [ [a4, [-15,-13,-11]] ] );
single piece groupoid: < a4, [ -15, -13, -11 ] >
gap> SetName( Ga4, "Ga4" );
gap> d := Arrow(Ga4, (1,3,4), -11, -13);
[(1,3,4) : -11 \rightarrow -13]
gap> aut1 := GroupoidAutomorphismByObjectPerm( Ga4, [-13,-11,-15] );;
gap> Display( aut1 );
 groupoid mapping: [ Ga4 ] -> [ Ga4 ]
root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]
images of objects: [ -13, -11, -15 ]
   images of rays: [(): -13 \rightarrow -13], (): -13 \rightarrow -11], (): -13 \rightarrow -15]
gap> d1 := ImageElm( aut1, d );
[(1,3,4) : -15 \rightarrow -11]
gap> h2 := GroupHomomorphismByImages( a4, a4, gensa4, [(2,3,4), (1,3,4)] );;
gap> aut2 := GroupoidAutomorphismByGroupAuto( Ga4, h2 );;
gap> Display( aut2 );
 groupoid mapping: [ Ga4 ] -> [ Ga4 ]
root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (2,3,4), (1,3,4) ] ]
images of objects: [ -15, -13, -11 ]
   images of rays: [ [() : -15 \rightarrow -15], [() : -15 \rightarrow -13], [() : -15 \rightarrow -11] ]
gap> d2 := ImageElm( aut2, d1 );
[(1,2,4) : -15 \rightarrow -11]
gap> im3 := [(), (1,3,2), (2,4,3)];;
gap> aut3 := GroupoidAutomorphismByRayShifts( Ga4, im3 );;
gap> Display( aut3 );
 groupoid mapping: [ Ga4 ] -> [ Ga4 ]
root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]
images of objects: [ -15, -13, -11 ]
   images of rays: [(): -15 \rightarrow -15], [(1,3,2): -15 \rightarrow -13],
```

```
[(2,4,3) : -15 -> -11] ]
gap> d3 := ImageElm( aut3, d2 );
[(1,4)(2,3) : -15 -> -11]
gap> d0 := Arrow( Ga4, (2,3,4), -11, -13 );;
gap> aut4 := GroupoidInnerAutomorphism( Ga4, d0 );;
gap> Display( aut4 );
groupoid mapping: [ Ga4 ] -> [ Ga4 ]
root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]
images of objects: [ -15, -11, -13 ]
  images of rays: [ [() : -15 -> -15], [(2,4,3) : -15 -> -11],
  [(2,3,4) : -15 -> -13] ]
gap> d4 := ImageElm( aut4, d3 );
[(1,2,4) : -15 -> -13]
```

### **5.6.2** Automorphisms of a groupoid with rays

Let S be a wide subgroupoid with rays of a standard groupoid G.

An automorphism of the root group extends to the whole of S with the rays fixed by the automorphism.

An automorphism of G by permuting the objects may map S to a different subgroupoid. So we construct an isomorphism  $\iota$  from S to a standard groupoid T, constgruct  $\alpha$  permuting the objects of T, and return  $\iota * \alpha * \iota^{-1}$ .

For an automorphism by ray shifts we require that the shifts are elements of the root group of S.

```
_ Example _
gap> s4c := Group( (1,2,3,4), (3,4));;
gap> SetName( s4c, "s4c" );
gap> s3c := Subgroup( s4c, [ (1,2), (2,3) ] );;
gap> SetName( s3c, "s3c" );
gap> Gs4c := SinglePieceGroupoid( s4c, [-9,-8,-7,-6] );;
gap> SetName( Gs4c, "Gs4c" );
gap> Hs3c := SubgroupoidWithRays( Gs4c, s3c, [ (), (1,4), (2,4), (3,4) ] );;
gap> SetName( Hs3c, "Hs3c" );
gap> ## (1) automorphism by group auto
gap> a1 := GroupHomomorphismByImages( s3c, s3c, [(1,2),(2,3)], [(1,3),(2,3)]);;
gap> aut1 := GroupoidAutomorphismByGroupAuto( Hs3c, a1 );
groupoid homomorphism : Hs3c -> Hs3c
[[(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4): -9 \rightarrow -8],
       [(2,4) : -9 \rightarrow -7], [(3,4) : -9 \rightarrow -6]],
  [(1,3): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4): -9 \rightarrow -8],
       [(2,4): -9 \rightarrow -7], [(3,4): -9 \rightarrow -6]]
gap> a := Arrow(Hs3c, (2,3,4), -8, -8);
[(2,3,4) : -8 \rightarrow -8]
gap> ImageElm( aut1, a );
[(2,4,3) : -8 \rightarrow -8]
gap>b := Arrow(Hs3c, (1,2,3,4), -7, -6);
[(1,2,3,4) : -7 \rightarrow -6]
gap> ## b = (2,4)(1,2)(3,4) \rightarrow (2,4)(1,3)(3,4)
gap> ImageElm( aut1, b );
[(1,4,2,3): -7 \rightarrow -6]
```

```
gap> ## (2) automorphism by object perm
gap> aut2 := GroupoidAutomorphismByObjectPerm( Hs3c, [-8,-7,-6,-9] );
groupoid homomorphism : Hs3c -> Hs3c
[[(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4): -9 \rightarrow -8],
       [(2,4): -9 \rightarrow -7], [(3,4): -9 \rightarrow -6]],
  [(2,4): -8 \rightarrow -8], [(2,3): -8 \rightarrow -8], [(1,2,4): -8 \rightarrow -7],
       [(1,3,4) : -8 \rightarrow -6], [(1,4) : -8 \rightarrow -9]]
gap> ImageElm( aut2, a );
[(1,4,3): -7 \rightarrow -7]
gap> ImageElm( aut2, b );
[(1,2)(3,4): -6 \rightarrow -9]
gap> ## (3) automorphism by ray shifts
gap> aut3 := GroupoidAutomorphismByRayShifts( Hs3c, [(),(2,3),(1,3),(1,2)] );
groupoid homomorphism : Hs3c -> Hs3c
[[(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4): -9 \rightarrow -8],
       [(2,4) : -9 \rightarrow -7], [(3,4) : -9 \rightarrow -6]],
  [(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4)(2,3): -9 \rightarrow -8],
       [(1,3)(2,4): -9 \rightarrow -7], [(1,2)(3,4): -9 \rightarrow -6]]
gap> ImageElm( aut3, a );
[(2,4,3): -8 \rightarrow -8]
gap> ImageElm( aut3, b );
[(1,4,2,3): -7 \rightarrow -6]
gap> e86 := Arrow( Hs3c, (1,3,2,4), -8, -6 );;
gap> aut86 := GroupoidInnerAutomorphism( Hs3c, e86 );
groupoid homomorphism : Hs3c -> Hs3c
[[(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(1,4): -9 \rightarrow -8],
       [(2,4): -9 \rightarrow -7], [(3,4): -9 \rightarrow -6]],
  [(1,2): -9 \rightarrow -9], [(2,3): -9 \rightarrow -9], [(2,4,3): -9 \rightarrow -6],
       [(2,4) : -9 \rightarrow -7], [(1,4)(2,3) : -9 \rightarrow -8]]
```

### 5.6.3 AutomorphismGroupOfGroupoid

```
▷ AutomorphismGroupOfGroupoid(gpd) (operation)
▷ NiceObjectAutoGroupGroupoid(gpd, aut) (operation)
```

As above, let G be the direct product of a group g and a complete digraph with n objects. The AutomorphismGroup  $\operatorname{Aut}(G)$  of G is isomorphic to the quotient of  $S_n \times A \times g^n$  by a subgroup isomorphic to g, where A is the automorphism group of g and  $S_n$  is the symmetric group on the n objects. This is one of the main topics in [AW10].

If H is the union of k groupoids, all isomorphic to G, then Aut(H) is isomorphic to  $S_k \ltimes Aut(G)$ .

The function NiceObjectAutoGroupGroupoid takes a groupoid and a subgroup of its automorphism group and retuns a *nice monomorphism* from this automorphism group to a pc-group, if one is available. The current implementation is experimental. Note that ImageElm at present only works on generating elements.

```
gap> AGa4 := AutomorphismGroupOfGroupoid( Ga4 );
Aut(Ga4)
gap> Length( GeneratorsOfGroup( AGa4 ) );
```

```
gap> AGgens := GeneratorsOfGroup( AGa4);;
gap> NGa4 := NiceObject( AGa4 );;
gap> MGa4 := NiceMonomorphism( AGa4 );;
gap> Size( AGa4 );
20736
gap> SetName( AGa4, "AGa4" );
gap> SetName( NGa4, "NGa4" );
gap> ## cannot test images of AGgens because of random variations
gap> ## Now do some tests!
gap> mgi := MappingGeneratorsImages( MGa4 );;
gap> autgen := mgi[1];;
gap> pcgen := mgi[2];;
gap> ngen := Length( autgen );;
gap> ForAll( [1..ngen], i -> Order(autgen[i]) = Order(pcgen[i]) );
true
```

### **5.6.4** Inner automorphisms

The inner automorphism subgroup  $\operatorname{Inn}(G)$  of the automorphism group of G is the group of inner automorphisms  $\land a : b \mapsto b^a$  for  $a \in G$ . It is *not* the case that the map  $G \to \operatorname{Inn}(G), a \mapsto \land a$  preserves multiplication. Indeed, when  $a = (o, g, p), b = (p, h, r) \in G$  with objects p, q, r all distict, then

$$\wedge (ab) = (\wedge a)(\wedge b)(\wedge a) = (\wedge b)(\wedge a)(\wedge b).$$

(Compare this with the permutation identity (pq)(qr)(pq) = (pr) = (qr)(pq)(qr).) So the map  $G \to \text{Inn}(G)$  is of type IsMappingWithObjectsByFunction.

In the example we convert the automorphism group AGa4 into a single object groupoid, and then define the inner automorphism map.

```
Example -
gap> AGa40 := Groupoid( AGa4, [0] );
single piece groupoid: < Aut(Ga4), [ 0 ] >
gap> conj := function(a)
             return ArrowNC( true, GroupoidInnerAutomorphism(Ga4,a), 0, 0 );
gap> inner := MappingWithObjectsByFunction( Ga4, AGa40, conj, [0,0,0] );;
gap> a1 := Arrow( Ga4, (1,2,3), -15, -13 );;
gap> inn1 := ImageElm( inner, a1 );;
gap> a2 := Arrow( Ga4, (2,3,4), -13, -11 );;
gap> inn2 := ImageElm( inner, a2 );;
gap> a3 := a1*a2;
[(1,3)(2,4): -15 \rightarrow -11]
gap> inn3 := ImageElm( inner, a3 );
[groupoid homomorphism : Ga4 \rightarrow Ga4
[[(1,2,3): -15 \rightarrow -15], [(2,3,4): -15 \rightarrow -15], [(): -15 \rightarrow -13],
      [(): -15 \rightarrow -11]],
  [[(1,3,4):-11->-11],[(1,2,4):-11->-11],[(1,3)(2,4):-11->-13],
      [() : -11 \rightarrow -15]] : 0 \rightarrow 0]
gap> (inn3 = inn1*inn2*inn1) and (inn3 = inn2*inn1*inn2);
```

true

### 5.6.5 GroupoidAutomorphismByGroupAutos

□ GroupoidAutomorphismByGroupAutos(gpd, auts)

(operation)

Homogeneous, discrete groupoids are the second type of groupoid for which a method is provided for AutomorphismGroupOfGroupoid. This is used in the XMod package for constructing crossed modules of groupoids. The two types of generating automorphism are GroupoidAutomorphismByGroupAutos, which requires a list of group automorphisms, one for each object group, and GroupoidAutomorphismByObjectPerm, which permutes the objects.

```
_ Example .
gap> Hs3 := HomogeneousDiscreteGroupoid( s3, [ -13..-10] );
homogeneous, discrete groupoid: < s3, [ -13 .. -10 ] >
gap> aut4 := GroupoidAutomorphismByObjectPerm( Hs3, [-12,-10,-11,-13] );
groupoid homomorphism : morphism from a homogeneous discrete groupoid:
[-13, -12, -11, -10] \rightarrow [-12, -10, -11, -13]
object homomorphisms:
IdentityMapping( s3 )
IdentityMapping( s3 )
IdentityMapping( s3 )
IdentityMapping( s3 )
gap> gens3 := GeneratorsOfGroup( s3 );;
gap> g1 := gens3[1];;
gap> g2 := gens3[2];;
gap> b1 := GroupHomomorphismByImages( s3, s3, gens3, [g1, g2^g1 ] );;
gap> b2 := GroupHomomorphismByImages( s3, s3, gens3, [g1^g2, g2 ] );;
gap> b3 := GroupHomomorphismByImages(s3, s3, gens3, [g1^2g2, g2^2(g1*g2)]);;
gap> b4 := GroupHomomorphismByImages( s3, s3, gens3, [g1^(g2*g1), g2^g1 ] );;
gap> aut5 := GroupoidAutomorphismByGroupAutos( Hs3, [b1,b2,b3,b4] );
groupoid homomorphism : morphism from a homogeneous discrete groupoid:
[-13, -12, -11, -10] \rightarrow [-13, -12, -11, -10]
object homomorphisms:
GroupHomomorphismByImages(s3,s3,
[(15,17,19)(16,18,20), (15,20)(16,19)(17,18)],
[(15,17,19)(16,18,20), (15,18)(16,17)(19,20)])
GroupHomomorphismByImages(s3,s3,
[(15,17,19)(16,18,20), (15,20)(16,19)(17,18)],
[(15,19,17)(16,20,18), (15,20)(16,19)(17,18)])
GroupHomomorphismByImages(s3,s3,
[(15,17,19)(16,18,20), (15,20)(16,19)(17,18)],
[(15,19,17)(16,20,18), (15,16)(17,20)(18,19)])
GroupHomomorphismByImages( s3, s3,
[(15,17,19)(16,18,20), (15,20)(16,19)(17,18)],
[(15,19,17)(16,20,18), (15,18)(16,17)(19,20)])
gap> genAHs3 := GeneratorsOfGroup( AHs3 );;
gap> Length( genAHs3 );
gap> ids3 := IdentityMapping( s3 );;
```

```
gap> aut5 := GroupoidAutomorphismByGroupAutos( Hs3, [b1^2,ids3,ids3,ids3] );;
gap> aut6 := GroupoidAutomorphismByGroupAutos( Hs3, [b2,ids3,ids3,ids3] );;
gap> aut7 := GroupoidAutomorphismByObjectPerm( Hs3, [ -12, -11, -10, -13 ] );;
gap> aut8 := GroupoidAutomorphismByObjectPerm( Hs3, [ -12, -13, -11, -10 ] );;
gap> ok := ForAll( genAHs3, a -> a in[ aut5, aut6, aut7, aut8 ] );
gap> nobAHs3 := NiceObject( AHs3 );;
gap> nmonAHs3 := NiceMonomorphism( AHs3 );;
gap> w := genAHs3[1];;
gap> w1 := ImageElm( nmonAHs3, w );;
gap> x := genAHs3[2];;
gap> x1 := ImageElm( nmonAHs3, x );;
gap> y := genAHs3[3];;
gap> y1 := ImageElm( nmonAHs3, y );;
gap> z := genAHs3[4];;
gap> z1 := ImageElm( nmonAHs3, z );;
gap> u := z*w*y*x*z;
groupoid homomorphism : morphism from a homogeneous discrete groupoid:
[-13, -12, -11, -10] \rightarrow [-11, -13, -10, -12]
object homomorphisms:
IdentityMapping( s3 )
ConjugatorAutomorphism( s3, (15,19,17)(16,20,18) )
IdentityMapping( s3 )
ConjugatorAutomorphism(s3, (15,20)(16,19)(17,18))
gap> u1 := z1*w1*y1*x1*z1;
(1,2,4,3)(5,17,23,16,8,20,26,13)(6,18,24,15,7,19,25,14)(9,21,27,12,10,22,28,
gap> imu := ImageElm( nmonAHs3, u );;
gap> u1 = imu;
true
```

### 5.6.6 AutomorphismGroupoidOfGroupoid

(attribute)

If G is a single piece groupoid with automorphism group  $\operatorname{Aut}(G)$ , and if H is the union of k pieces, all isomorphic to G, then the automorphism group of H is the wreath product  $S_k \ltimes \operatorname{Aut}(G)$ . However, we find it more convenient to construct the *automorphism groupoid* of H. This is a single piece groupoid  $\operatorname{Aut}(H)$  with k objects – the object lists of the pieces of H – and root group  $\operatorname{Aut}(G)$ . Isomorphisms between the root groups of the k pieces may be applied to the generators of  $\operatorname{Aut}(G)$  to construct automorphism groups of these pieces, and then isomorphisms between these automorphism groups. We then construct  $\operatorname{AUT}(H)$  using GroupoidByIsomorphisms.

In the special case that H is homogeneous, there is no need to construct a collection of automorphism groups. Rather, the rays of AUT(H) are given by IsomorphismNewObjects. For the example we use Hd8 constructed in subsection HomogeneousGroupoid (4.1.5).

# 5.7 Matrix representations of groupoids

Suppose that gpd is the direct product of a group G and a complete digraph, and that  $\rho: G \to M$  is an isomorphism to a matrix group M. Then if rep is the isomorphic groupoid with the same objects and root group M there is an isomorphism  $\mu$  from gpd to rep mapping  $(g: i \to j)$  to  $(\rho g: i \to j)$ .

When gpd is a groupoid with rays, a representation can be obtained by restricting a representation of its parent.

```
Example .
gap> reps := IrreducibleRepresentations( s4 );;
gap> rep4 := reps[4];;
gap> Rs4 := Groupoid( Image( rep4 ), ObjectList( Gs4 ) );
single piece groupoid: < Group([ [ [ 0, 1, 0 ], [ 1, 0, 0 ], [ 0, 0, 1 ] ],
  [[0,0,1],[1,0,0],[0,1,0]],
  [[-1, 0, 0], [0, 1, 0], [0, 0, -1]],
  [[1, 0, 0], [0, -1, 0], [0, 0, -1]]]), [-15, -14, -13, -12, -11]
 ] >
gap> IsMatrixGroupoid( Rs4 );
true
gap> gens := GeneratorsOfGroupoid( Gs4 );
[(1,2,3,4): -15 \rightarrow -15], [(3,4): -15 \rightarrow -15], [(): -15 \rightarrow -14],
  [(): -15 \rightarrow -13], [(): -15 \rightarrow -12], [(): -15 \rightarrow -11]]
gap> images := List( gens,
         g -> Arrow( Rs4, ImageElm(rep4,g![1]), g![2], g![3] ) );
[[[[-1, 0, 0], [0, 0, 1], [0, -1, 0]] : -15 \rightarrow -15],
  [[[0, 1, 0], [1, 0, 0], [0, 0, 1]] : -15 \rightarrow -15],
  [[[1, 0, 0], [0, 1, 0], [0, 0, 1]] : -15 \rightarrow -14],
  [[[1, 0, 0], [0, 1, 0], [0, 0, 1]] : -15 \rightarrow -13],
  [[[1, 0, 0], [0, 1, 0], [0, 0, 1]] : -15 \rightarrow -12],
  [[[1, 0, 0], [0, 1, 0], [0, 0, 1]] : -15 \rightarrow -11]]
gap> mor := GroupoidHomomorphismFromSinglePiece( Gs4, Rs4, gens, images );
groupoid homomorphism :
[[(1,2,3,4): -15 \rightarrow -15], [(3,4): -15 \rightarrow -15], [(): -15 \rightarrow -14],
```

```
[(): -15 \rightarrow -13], [(): -15 \rightarrow -12], [(): -15 \rightarrow -11]],
  [[[[-1, 0, 0], [0, 0, 1], [0, -1, 0]] : -15 \rightarrow -15],
      [[[0, 1, 0], [1, 0, 0], [0, 0, 1]] : -15 \rightarrow -15],
      [[[1, 0, 0], [0, 1, 0], [0, 0, 1]] : -15 \rightarrow -14],
      [[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 \rightarrow -13],
      [[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 \rightarrow -12],
      [[[1,0,0],[0,1,0],[0,0,1]]:-15->-11]]]
gap> a := Arrow( Hs4, (1,4,2), -12, -13);
[(1,4,2) : -12 \rightarrow -13]
gap> ImageElm( mor, a );
[[[0, 0, -1], [-1, 0, 0], [0, 1, 0]] : -12 \rightarrow -13]
gap> rmor := RestrictedMappingGroupoids( mor, Hd8b );
groupoid homomorphism :
 \hbox{\tt [[(1,2,3,4):-14->-14],[(1,3):-14->-14],[(1,2,3):-14->-13],} 
      [(1,2,4) : -14 \rightarrow -12]],
  [[[[-1, 0, 0], [0, 0, 1], [0, -1, 0]] : -14 \rightarrow -14],
      [[[1, 0, 0], [0, 0, -1], [0, -1, 0]] : -14 \rightarrow -14],
      [[[0, 0, 1], [-1, 0, 0], [0, -1, 0]] : -14 \rightarrow -13],
      [[[0, -1, 0], [0, 0, 1], [-1, 0, 0]] : -14 \rightarrow -12]]
```

# Chapter 6

# **Graphs of Groups and Groupoids**

This package was originally designed to implement *graphs of groups*, a notion introduced by Serre in [Ser80]. It was only when this was extended to *graphs of groupoids* that the functions for groupoids, described in the previous chapters, were required. The methods described here are based on Philip Higgins' paper [Hig76]. For further details see Chapter 2 of [Moo01]. Since a graph of groups involves a directed graph, with a group associated to each vertex and arc, we first define digraphs with edges weighted by the generators of a free group.

# 6.1 Digraphs

### 6.1.1 FpWeightedDigraph

```
    ▷ FpWeightedDigraph(verts, arcs)
    ▷ IsFpWeightedDigraph(dig)
    ▷ InvolutoryArcs(dig)
    (attribute)
    (attribute)
```

A weighted digraph is a record with two components: vertices, which are usually taken to be positive integers (to distinguish them from the objects in a groupoid); and arcs, which take the form of 3-element lists [weight,tail,head]. The tail and head are the two vertices of the arc. The weight is taken to be an element of a finitely presented group, so as to produce digraphs of type IsFpWeightedDigraph.

```
gap> V1 := [ 5, 6 ];;
gap> fg1 := FreeGroup( "y" );;
gap> y := fg1.1;;
gap> A1 := [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ];
gap> D1 := FpWeightedDigraph( fg1, V1, A1 );
weighted digraph with vertices: [ 5, 6 ]
and arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]
gap> inv1 := InvolutoryArcs( D1 );
[ 2, 1 ]
```

The example illustrates the fact that we require arcs to be defined in involutory pairs, as though they were inverse elements in a groupoid. We may in future decide just to give [y,5,6] as the data and

get the function to construct the reverse edge. The attribute InvolutoryArcs returns a list of the positions of each inverse arc in the list of arcs. In the second example the graph is a complete digraph on three vertices.

# **6.2** Graphs of Groups

### 6.2.1 GraphOfGroups

A graph of groups is traditionally defined as consisting of:

- a digraph with involutory pairs of arcs;
- a vertex group associated to each vertex;
- a group associated to each pair of arcs;
- an injective homomorphism from each arc group to the group at the head of the arc.

We have found it more convenient to associate to each arc:

- a subgroup of the vertex group at the tail;
- a subgroup of the vertex group at the head;
- an isomorphism between these subgroups, such that each involutory pair of arcs determines inverse isomorphisms.

These two viewpoints are clearly equivalent.

In this implementation we require that all subgroups are of finite index in the vertex groups.

The three attributes provide a means of calling the three items of data in the construction of a graph of groups.

We shall be representing free products with amalgamation of groups and HNN extensions of groups in Section 6.4. So we take as our first example the trefoil group with generators a, b and

relation  $a^3 = b^2$ . For this we take digraph D1 above with an infinite cyclic group at each vertex, generated by a and b respectively. The two subgroups will be generated by  $a^3$  and  $b^2$  with the obvious isomorphisms.

```
_ Example _
gap> ## free vertex group at 5
gap> fa := FreeGroup( "a" );;
gap> a := fa.1;;
gap> SetName( fa, "fa" );
gap> hy := Subgroup( fa, [a^3] );;
gap> SetName( hy, "hy" );
gap> ## free vertex group at 6
gap> fb := FreeGroup( "b" );;
gap> b := fb.1;;
gap> SetName( fb, "fb" );
gap> hybar := Subgroup( fb, [b^2] );;
gap> SetName( hybar, "hybar" );
gap> ## isomorphisms between subgroups
gap> homy := GroupHomomorphismByImagesNC( hy, hybar, [a^3], [b^2] );;
gap> homybar := GroupHomomorphismByImagesNC( hybar, hy, [b^2], [a^3] );;
gap> ## defining graph of groups G1
gap> G1 := GraphOfGroups( D1, [fa,fb], [homy,homybar] );
Graph of Groups: 2 vertices; 2 arcs; groups [fa, fb]
gap> Display( G1 );
Graph of Groups with :-
    vertices: [ 5, 6 ]
        arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]
      groups: [ fa, fb ]
isomorphisms: [ [ [ a^3 ], [ b^2 ] ], [ [ b^2 ], [ a^3 ] ] ]
```

### 6.2.2 IsGraphOfFpGroups

```
    ▷ IsGraphOfFpGroups(gg)
    ▷ IsGraphOfPcGroups(gg)
    ▷ IsGraphOfPermGroups(gg)
    ○ (property)
    ○ (property)
```

This is a list of properties to be expected of a graph of groups. In principle any type of group known to GAP may be used as vertex groups, though these types are not normally mixed in a single structure.

```
gap> IsGraphOfFpGroups( G1 );
true
gap> IsomorphismsOfGraphOfGroups( G1 );
[ [ a^3 ] -> [ b^2 ] , [ b^2 ] -> [ a^3 ] ]
```

### 6.2.3 RightTransversalsOfGraphOfGroups

Computation with graph of groups words will require, for each arc subgroup ha, a set of representatives for the left cosets of ha in the tail vertex group. As already pointed out, we require subgroups of finite index. Since GAP prefers to provide right cosets, we obtain the right representatives first, and then invert them.

When the vertex groups are of type FpGroup we shall require normal forms for these groups, so we assume that such vertex groups are provided with Knuth Bendix rewriting systems using functions from the main GAP library, (e.g. IsomorphismFpSemigroup).

```
gap> RTG1 := RightTransversalsOfGraphOfGroups( G1 );
[ [ <identity ...>, a, a^2 ], [ <identity ...>, b ] ]
gap> LTG1 := LeftTransversalsOfGraphOfGroups( G1 );
[ [ <identity ...>, a^-1, a^-2 ], [ <identity ...>, b^-1 ] ]
```

## 6.3 Words in a Graph of Groups and their normal forms

## 6.3.1 GraphOfGroupsWord

```
      ▷ GraphOfGroupsWord(gg, tv, list)
      (operation)

      ▷ IsGraphOfGroupsWord(w)
      (property)

      ▷ GraphOfGroupsOfWord(w)
      (attribute)

      ▷ WordOfGraphOfGroupsWord(w)
      (attribute)

      ▷ TailOfGraphOfGroupsWord(w)
      (attribute)

      ▷ HeadOfGraphOfGroupsWord(w)
      (attribute)
```

If G is a graph of groups with underlying digraph D, the following groupoids may be considered. First there is the free groupoid or path groupoid on D. Since we want each involutory pair of arcs to represent inverse elements in the groupoid, we quotient out by the relations  $y^-1 = y$ bar to obtain PG(D). Secondly, there is the discrete groupoid VG(D), namely the union of all the vertex groups. Since these two groupoids have the same object set (the vertices of D) we can form A(G), the free product of PG(D) and VG(D) amalgamated over the vertices. For further details of this universal groupoid construction see [Moo01]. (Note that these groupoids are not implemented in this package.)

An element of A(G) is a graph of groups word which may be represented by a list of the form  $w = [g_1, y_1, g_2, y_2, ..., g_n, y_n, g_{n+1}]$ . Here each  $y_i$  is an arc of D; the head of  $y_{i-1}$  is a vertex  $v_i$  which is also the tail of  $y_i$ ; and  $g_i$  is an element of the vertex group at  $v_i$ .

So a graph of groups word requires as data the graph of groups; the tail vertex for the word; and a list of arcs and group elements. We may specify each arc by its position in the list of arcs.

In the following example, where gw1 is a word in the trefoil graph of groups, the  $y_i$  are specified by their positions in A1. Both arcs are traversed twice, so the resulting word is a loop at vertex 5.

```
gap> L1 := [ a^7, 1, b^-6, 2, a^-11, 1, b^9, 2, a^7 ];;
```

```
gap> gw1 := GraphOfGroupsWord( G1, 5, L1 );
(5)a^7.y.b^-6.y^-1.a^-11.y.b^9.y^-1.a^7(5)
gap> IsGraphOfGroupsWord( gw1 );
true
gap> [ TailOfGraphOfGroupsWord(gw1), HeadOfGraphOfGroupsWord(gw1) ];
[ 5, 5 ]
gap> GraphOfGroupsOfWord(gw1);
Graph of Groups: 2 vertices; 2 arcs; groups [ fa, fb ]
gap> WordOfGraphOfGroupsWord( gw1 );
[ a^7, 1, b^-6, 2, a^-11, 1, b^9, 2, a^7 ]
```

### 6.3.2 ReducedGraphOfGroupsWord

A graph of groups word may be reduced in two ways, to give a normal form. Firstly, if part of the word has the form [yi, identity, yibar] then this subword may be omitted. This is known as a length reduction. Secondly there are coset reductions. Working from the left-hand end of the word, subwords of the form  $[g_i, y_i, g_{i+1}]$  are replaced by  $[t_i, y_i, m_i(h_i) * g_{i+1}]$  where  $g_i = t_i * h_i$  is the unique factorisation of  $g_i$  as a left coset representative times an element of the arc subgroup, and  $m_i$  is the isomorphism associated to  $y_i$ . Thus we may consider a coset reduction as passing a subgroup element along an arc. The resulting normal form (if no length reductions have taken place) is then  $[t_1, y_1, t_2, y_2, ..., t_n, y_n, k]$  for some k in the head group of  $y_n$ . For further details see Section 2.2 of [Moo01].

The reduction of the word gw1 in our example includes one length reduction. The four stages of the reduction are as follows:

```
a^{7}b^{-6}a^{-11}b^{9}a^{7} \mapsto a^{-2}b^{0}a^{-11}b^{9}a^{7} \mapsto a^{-13}b^{9}a^{7} \mapsto a^{-1}b^{-8}b^{9}a^{7} \mapsto a^{-1}b^{-1}a^{10}.
gap> nw1 := ReducedGraphOfGroupsWord( gw1 );
(5)a^{-1}.y.b^{-1}.y^{-1}.a^{10}(5)
```

### 6.4 Free products with amalgamation and HNN extensions

### 6.4.1 FreeProductWithAmalgamation

```
▷ FreeProductWithAmalgamation(gp1, gp2, iso)

▷ FreeProductWithAmalgamationInfo(fpa)

▷ IsFreeProductWithAmalgamation(fpa)

▷ GraphOfGroupsRewritingSystem(fpa)

▷ NormalFormGGRWS(fpa, word)

(operation)

(attribute)

(attribute)

(attribute)
```

As we have seen with the trefoil group example in Section 6.2, graphs of groups can be used to obtain a normal form for free products with amalgamation  $G_1 *_H G_2$  when  $G_1, G_2$  both have rewrite systems, and H is of finite index in both  $G_1$  and  $G_2$ .

When gp1 and gp2 are fp-groups, the operation FreeProductWithAmalgamation constructs the required fp-group. When the two groups are permutation groups, the IsomorphismFpGroup operation is called on both gp1 and gp2, and the resulting isomorphism is transported to one between the two new subgroups.

The attribute GraphOfGroupsRewritingSystem of fpa is the graph of groups which has underlying digraph D1, with two vertices and two arcs; the two groups as vertex groups; and the specified isomorphisms on the arcs. Despite the name, graphs of groups constructed in this way *do not* belong to the category IsRewritingSystem. This anomaly may be dealt with when time permits.

The example below shows a computation in the the free product of the symmetric s3 and the alternating a4, amalgamated over a cyclic subgroup c3.

```
_ Example
gap> ## set up the first group s3 and a subgroup c3=<a1>
gap> fg2 := FreeGroup( 2, "a" );;
gap > rel1 := [fg2.1^3, fg2.2^2, (fg2.1*fg2.2)^2];;
gap> s3 := fg2/rel1;;
gap> gs3 := GeneratorsOfGroup(s3);;
gap> SetName( s3, "s3" );
gap> a1 := gs3[1];; a2 := gs3[2];;
gap> H1 := Subgroup(s3,[a1]);;
gap> ## then the second group a4 and subgroup c3=<b1>
gap> f2 := FreeGroup( 2, "b" );;
gap> rel2 := [ f2.1^3, f2.2^3, (f2.1*f2.2)^2 ];;
gap> a4 := f2/rel2;;
gap> ga4 := GeneratorsOfGroup(a4);;
gap> SetName( a4, "a4" );
gap> b1 := ga4[1]; b2 := ga4[2];;
gap> H2 := Subgroup(a4,[b1]);;
gap> ## form the isomorphism and the fpa group
gap> iso := GroupHomomorphismByImages(H1,H2,[a1],[b1]);;
gap> inv := InverseGeneralMapping(iso);;
gap> fpa := FreeProductWithAmalgamation( s3, a4, iso );
<fp group on the generators [ f1, f2, f3, f4 ]>
gap> RelatorsOfFpGroup( fpa );
[ f1^2, f2^3, (f2*f1)^2, f3^3, f4^3, (f4*f3)^2, f2*f3^-1 ]
gap> gg1 := GraphOfGroupsRewritingSystem( fpa );;
gap> Display( gg1 );
Graph of Groups with :-
    vertices: [ 5, 6 ]
        arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]
      groups: [ s3, a4 ]
isomorphisms: [ [ [ a1 ], [ b1 ] ], [ [ b1 ], [ a1 ] ]
gap> LeftTransversalsOfGraphOfGroups( gg1 );
[ [ <identity ...>, a2^-1 ], [ <identity ...>, b2^-1, b1^-1*b2^-1, b1*b2^-1 ]
gap> gfpa := GeneratorsOfGroup( fpa );;
gap> w2 := (gfpa[1]*gfpa[2]*gfpa[3]^gfpa[4])^3;
(f1*f2*f4^-1*f3*f4)^3
gap> n2 := NormalFormGGRWS( fpa, w2 );
f2*f3*(f4^-1*f2)^2*f4^-1*f3
```

### 6.4.2 ReducedImageElm

```
▷ ReducedImageElm(hom, eml) (operation)
▷ IsMappingToGroupWithGGRWS(map) (property)
▷ Embedding(fpa, num) (method)
```

All fpa-groups are provided with a record attribute, FreeProductWithAmalgamationInfo(fpa) which is a record storing the groups, subgroups and isomorphism involved in their construction. This information record also contains the embeddings of the two groups into the product. The operation ReducedImageElm, applied to a homomorphism h of type IsMappingToGroupWithGGRWS and an element x of the source, finds the usual ImageElm(h,x) and then reduces this to its normal form using the graph of groups rewriting system.

```
gap> fpainfo;
rec( embeddings := [ [ a2, a1 ] -> [ f1, f2 ], [ b1, b2 ] -> [ f3, f4 ] ],
    groups := [ s3, a4 ], isomorphism := [ a1 ] -> [ b1 ],
    positions := [ [ 1, 2 ], [ 3, 4 ] ],
    subgroups := [ Group([ a1 ]), Group([ b1 ]) ] )
gap> emb2 := Embedding( fpa, 2 );
[ b1, b2 ] -> [ f3, f4 ]
gap> ImageElm( emb2, b1^b2 );
f4^-1*f3*f4
gap> ReducedImageElm( emb2, b1^b2 );
f4*f3^-1
```

#### 6.4.3 HnnExtension

```
▷ HnnExtension(gp, iso) (operation)
▷ HnnExtensionInfo(gp, iso) (attribute)
▷ IsHnnExtension(hnn) (property)
```

For *HNN extensions*, the appropriate graph of groups has underlying digraph with just one vertex and one pair of loops, weighted with FpGroup generators  $z, z^{-1}$ . There is one vertex group G, two isomorphic subgroups H1, H2 of G, with the isomorphism and its inverse on the loops. The presentation of the extension has one more generator than that of G and corresponds to the generator z.

The functions GraphOfGroupsRewritingSystem and NormalFormGGRWS may be applied to hnn-groups as well as to fpa-groups.

In the example we take G=a4 and the two subgroups are cyclic groups of order 3.

```
gap> H3 := Subgroup(a4,[b2]);;
gap> i23 := GroupHomomorphismByImages( H2, H3, [b1], [b2] );;
gap> hnn := HnnExtension( a4, i23 );
<fp group on the generators [ fe1, fe2, fe3 ]>
gap> phnn := PresentationFpGroup( hnn );;
gap> TzPrint( phnn );
#I generators: [ fe1, fe2, fe3 ]
```

```
#I relators:
#I 1. 3 [ 1, 1, 1 ]
#I 2. 3 [ 2, 2, 2 ]
#I 3. 4 [ 1, 2, 1, 2 ]
#I 4. 4 [ -3, 1, 3, -2 ]
gap> gg2 := GraphOfGroupsRewritingSystem( hnn );
Graph of Groups: 1 vertices; 2 arcs; groups [ a4 ]
gap> LeftTransversalsOfGraphOfGroups( gg2 );
[ [ <identity ...>, b2^-1, b1^-1*b2^-1, b1*b2^-1 ],
        [ <identity ...>, b1^-1, b1, b2^-1*b1 ] ]
gap> gh := GeneratorsOfGroup( hnn );
gap> w3 := (gh[1]^gh[2])*gh[3]^-1*(gh[1]*gh[3]*gh[2]^2)^2*gh[3]*gh[2];
fe2^-1*fe1*fe2*fe3^-1*(fe1*fe3*fe2^2)^2*fe3*fe2
gap> n3 := NormalFormGGRWS( hnn, w3 );
(fe2*fe1*fe3)^2
```

As with fpa-groups, hnn-groups are provided with a record attribute, HnnExtensionInfo(hnn), storing the group, subgroups and isomorphism involved in their construction.

```
gap> hnninfo := HnnExtensionInfo( hnn );
rec( embeddings := [ [ b1, b2 ] -> [ fe1, fe2 ] ], group := a4,
   isomorphism := [ b1 ] -> [ b2 ],
   subgroups := [ Group([ b1 ]), Group([ b2 ]) ] )
gap> emb := Embedding( hnn, 1 );
[ b1, b2 ] -> [ fe1, fe2 ]
gap> ImageElm( emb, b1^b2 );
fe2^-1*fe1*fe2
gap> ReducedImageElm( emb, b1^b2 );
fe2*fe1^-1
```

# 6.5 GraphsOfGroupoids and their Words

#### 6.5.1 GraphOfGroupoids

```
▷ GraphOfGroupoids(dig, gpds, subgpds, isos)
                                                                             (operation)
▷ IsGraphOfPermGroupoids(gg)
                                                                              (property)
▷ IsGraphOfFpGroupoids(gg)
                                                                              (property)

    □ GroupoidsOfGraphOfGroupoids(gg)

                                                                              (attribute)
▷ DigraphOfGraphOfGroupoids(gg)
                                                                              (attribute)
(attribute)
▷ IsomorphismsOfGraphOfGroupoids(gg)
                                                                              (attribute)
▷ RightTransversalsOfGraphOfGroupoids(gg)
                                                                              (attribute)
▷ LeftTransversalsOfGraphOfGroupoids(gg)
                                                                              (attribute)
```

Graphs of groups generalise naturally to graphs of groupoids, forming the class IsGraphOfGroupoids. There is now a groupoid at each vertex and the isomorphism on an arc identifies wide subgroupoids at the tail and at the head. Since all subgroupoids are wide, every groupoid in

a connected constituent of the graph has the same number of objects, but there is no requirement that the object sets are all the same.

The example below generalises the trefoil group example in subsection 4.4.1, taking at each vertex of D1 a two-object groupoid with a free group on one generator, and full subgroupoids with groups  $\langle a^3 \rangle$  and  $\langle b^2 \rangle$ .

```
__ Example _
gap> Gfa := SinglePieceGroupoid( fa, [-2,-1] );;
gap> SetName( Gfa, "Gfa" );
gap> Uhy := Subgroupoid( Gfa, [ [ hy, [-2,-1] ] ] );;
gap> SetName( Uhy, "Uhy" );
gap> Gfb := SinglePieceGroupoid( fb, [-4,-3] );;
gap> ofa := One( fa );; ofb := One( fb );;
gap> SetName( Gfb, "Gfb" );
gap> Uhybar := Subgroupoid( Gfb, [ [ hybar, [-4,-3] ] ] );;
gap> SetName( Uhybar, "Uhybar" );
gap> gens := GeneratorsOfGroupoid( Uhy );;
gap> gensbar := GeneratorsOfGroupoid( Uhybar );;
gap> mory := GroupoidHomomorphismFromSinglePiece(
                 Uhy, Uhybar, gens, gensbar );
groupoid homomorphism : Uhy -> Uhybar
[ [ [a^3 : -2 \rightarrow -2], [<identity ...> : -2 \rightarrow -1] ],
  [ [b^2 : -4 \rightarrow -4], [<identity ...> : -4 -> -3] ] ]
gap> morybar := InverseGeneralMapping( mory );
groupoid homomorphism : Uhybar -> Uhy
[ [ [b^2 : -4 -> -4], [<identity ...> : -4 -> -3] ],
  [ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ] ]
gap> gg3 := GraphOfGroupoids( D1, [Gfa,Gfb], [Uhy,Uhybar], [mory,morybar] );;
gap> Display( gg3 );
Graph of Groupoids with :-
    vertices: [5, 6]
        arcs: [[y, 5, 6], [y^-1, 6, 5]]
   groupoids:
fp single piece groupoid: Gfa
  objects: [ -2, -1 ]
    group: fa = <[ a ]>
fp single piece groupoid: Gfb
  objects: [ -4, -3 ]
    group: fb = <[ b ]>
subgroupoids: single piece groupoid: Uhy
  objects: [ -2, -1 ]
    group: hy = <[a^3]>
single piece groupoid: Uhybar
  objects: [ -4, -3 ]
    group: hybar = <[ b^2 ]>
isomorphisms: [ groupoid homomorphism : Uhy -> Uhybar
    [ [ [a^3 : -2 \rightarrow -2], [<identity ...> : -2 -> -1] ],
      [ [b^2 : -4 \rightarrow -4], [<identity ...> : -4 -> -3] ] ],
  groupoid homomorphism : Uhybar -> Uhy
    [ [ [b^2 : -4 \rightarrow -4], [<identity ...> : -4 -> -3] ],
      [ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ] ] ]
```

### 6.5.2 GraphOfGroupoidsWord

```
▷ GraphOfGroupoidsWord(gg, tv, list) (operation)
▷ IsGraphOfGroupoidsWord(w) (property)
▷ GraphOfGroupoidsOfWord(w) (attribute)
▷ WordOfGraphOfGroupoidsWord(w) (operation)
▷ ReducedGraphOfGroupoidsWord(w) (operation)
▷ IsReducedGraphOfGroupoidsWord(w) (property)
```

Having produced the graph of groupoids gg3, we may construct left coset representatives; choose a graph of groupoids word; and reduce this to normal form. Compare the nw3 below with the normal form nw1 in subsection 4.3.2.

The reduction proceeds as follows.

```
• [a^7:-1 \to -2] = [a^{-2}:-1 \to -1] * [a^9:-1 \to -2] \xrightarrow{y} [a^{-2}:-1 \to -1] * [b^6:-3 \to -4]
```

• 
$$[b^6: -3 \to -4] * [b^{-6}: -4 \to -4] = [id: -3 \to -4] \xrightarrow{\bar{y}} [id: -1 \to -2]$$

• 
$$[a^{-2}:-1 \to -1] * [\mathrm{id}:-1 \to -2] * [a^{-11}:-2 \to -1] = [a^{-13}:-1 \to -1]$$

$$\bullet \ [a^{-13}:-1\to -1] = [a^{-1}:-1\to -1] * [a^{-12}:-1\to -1] \xrightarrow{y} [a^{-1}:-1\to -1] * [b^{-8}:-3\to -3]$$

• 
$$[b^{-8}: -3 \to -3] * [b^9: -3 \to -4] = [b^{-1}: -3 \to -3] * [b^2: -3 \to -4] \xrightarrow{\bar{y}} [b^{-1}: -3 \to -3] * [a^3: -1 \to -2]$$

• 
$$[a^3 := -1 \to -2] * [a^7 : -2 \to -1] = [a^{10} : -1 \to -1]$$

# Chapter 7

# **Technical Notes**

This short chapter is included for the benefit of anyone wishing to implement some other variety of many-object structures, for example *ringoids*, which are rings with many objects; *Lie groupoids*, which are Lie groups with many objects; and so on.

# 7.1 Many object structures

Structures with many objects, and their elements, are defined in a manner similar to the single object case. For elements we have:

```
    DeclareCategory( "IsMultiplicativeElementWithObjects",
IsMultiplicativeElement );
```

- DeclareCategory( "IsMultiplicativeElementWithObjectsAndOnes", IsMultiplicativeElementWithObjects );
- DeclareCategory( "IsMultiplicativeElementWithObjectsAndInverses", IsMultiplicativeElementWithObjectsAndOnes );
- DeclareCategory( "IsGroupoidElement",
   IsMultiplicativeElementWithObjectsAndInverses );

as well as various category collections. For the various structures we have:

- DeclareCategory( "IsDomainWithObjects", IsDomain );
- DeclareCategory( "IsMagmaWithObjects", IsDomainWithObjects and IsMultiplicativeElementWithObjectsCollection );
- DeclareCategory( "IsSemigroupWithObjects", IsMagmaWithObjects and IsAssociative );
- DeclareCategory( "IsMonoidWithObjects", IsSemigroupWithObjects and IsMultiplicativeElementWithObjectsAndOnesCollection );
   IsMultiplicativeElementWithObjectsAndInversesCollection );
- DeclareCategory( "IsGroupoid", IsMonoidWithObjects and IsGroupoidElementCollection );

Among the groupoids constructed earlier are the single piece Gd8 and the five component union U5:

```
_ Example
gap> CategoriesOfObject( Gd8 );
[ "IsListOrCollection", "IsCollection", "IsExtLElement",
  "CategoryCollections(IsExtLElement)", "IsExtRElement",
  "CategoryCollections(IsExtRElement)",
  "CategoryCollections(IsMultiplicativeElement)", "IsGeneralizedDomain",
  "IsMagma", "IsDomainWithObjects",
  "CategoryCollections(IsMultiplicativeElementWithObjects)",
  "CategoryCollections(IsMultiplicativeElementWithObjectsAndOnes)",
  "CategoryCollections(IsMultiplicativeElementWithObjectsAndInverses)
", "CategoryCollections(IsGroupoidElement)", "IsMagmaWithObjects",
  "IsMagmaWithObjectsAndOnes", "IsMagmaWithObjectsAndInverses",
  "IsGroupoid" ]
gap> FamilyObj( Gd8 );
                         ## these numbers vary from one run to another
NewFamily( "GroupoidFamily", [ 2722 ], [ 53, 54, 79, 80, 81, 82, 92, 93, 116,
  117, 119, 120, 123, 205, 501, 2690, 2703, 2707, 2711, 2715, 2718, 2720,
  2721, 2722 ] )
gap> KnownAttributesOfObject( Gd8 );
[ "Name", "Size", "ObjectList", "GeneratorsOfMagmaWithObjects",
  "GeneratorsOfGroupoid" ]
gap> KnownTruePropertiesOfObject( Gd8 );
[ "IsNonTrivial", "IsFinite", "IsDuplicateFree", "IsAssociative",
  "IsSinglePieceDomain", "IsDirectProductWithCompleteDigraphDomain"]
gap> RepresentationsOfObject( Gd8 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsMWOSinglePieceRep" ]
gap> RepresentationsOfObject( U5 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPiecesRep" ]
```

Similarly, for arrows, we have:

```
_ Example __
gap> [ a78, e2 ];
[m2:-7 \rightarrow -8], [(1,3):-8 \rightarrow -7]]
gap> CategoriesOfObject(a78);
[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",
  "IsMultiplicativeElementWithObjects" ]
gap> FamilyObj( a78 ); ## again these numbers vary
NewFamily( "MultiplicativeElementWithObjectsFamily", [ 2702 ],
[ 79, 80, 81, 82, 116, 119, 122, 2702 ] )
gap> CategoriesOfObject(e2);
[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",
  "IsMultiplicativeElementWithObjects",
  "IsMultiplicativeElementWithObjectsAndOnes",
  "IsMultiplicativeElementWithObjectsAndInverses",
  "IsGroupoidElement" ]
gap> FamilyObj( e2 );
NewFamily( "GroupoidElementFamily", [ 2714 ],
[ 79, 80, 81, 82, 116, 119, 122, 2702, 2706, 2710, 2714 ] )
```

## 7.2 Many object homomorphisms

Homomorphisms of structures with many objects have a similar heirarchy. A few examples:

- DeclareCategory( "IsGeneralMappingWithObjects", IsGeneralMapping );
- DeclareSynonymAttr( "IsMagmaWithObjectsGeneralMapping", IsGeneralMappingWithObjects and RespectsMultiplication );
- DeclareSynonymAttr( "IsMagmaWithObjectsHomomorphism", IsMagmaWithObjectsGeneralMapping and IsMapping );
- DeclareCategory("IsGroupoidHomomorphism", IsMagmaWithObjectsHomomorphism);

Two forms of representation are used: for mappings to a single piece; and for unions of such mappings:

```
• DeclareRepresentation( "IsMappingToSinglePieceRep",
IsMagmaWithObjectsHomomorphism and IsAttributeStoringRep and
IsGeneralMapping, [ "Source", "Range", "SinglePieceMappingData" ] );
```

```
• DeclareRepresentation( "IsMappingWithObjectsRep",
IsMagmaWithObjectsHomomorphism and IsAttributeStoringRep and
IsGeneralMapping, [ "Source", "Range", "PiecesOfMapping" ] );
```

In previous chapters, hom1 was an endofunction on M78; homd8 was a homomorphism from Gd8 to Gs3; and aut3 was an automorphism of Ga4. All homomorphisms have family GeneralMappingWithObjectsFamily. Perhaps it would be better to have separate families for each structure?

```
_ Example .
gap> FamilyObj(hom1);
NewFamily( "GeneralMappingWithObjectsFamily", [ 2726 ],
[ 79, 80, 81, 82, 116, 119, 122, 126, 130, 149, 412, 2726 ] )
gap> KnownAttributesOfObject( hom1 );
[ "Range", "Source", "SinglePieceMappingData" ]
gap> KnownTruePropertiesOfObject( hom1 );
[ "CanEasilyCompareElements", "CanEasilySortElements", "IsTotal",
  "IsSingleValued", "RespectsMultiplication", "IsGeneralMappingToSinglePiece",
  "IsGeneralMappingFromSinglePiece", "IsInjectiveOnObjects",
  "IsSurjectiveOnObjects" ]
gap> CategoriesOfObject( homd8 );
[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",
  "IsMultiplicativeElementWithOne", "IsMultiplicativeElementWithInverse",
  "IsAssociativeElement", "IsGeneralMapping", "IsGeneralMappingWithObjects",
  "IsGroupoidHomomorphism" ]
gap> KnownAttributesOfObject( homd8 );
[ "Range", "Source", "SinglePieceMappingData", "ImagesOfObjects", "ImageElementsOfRays",
  "ObjectTransformationOfGroupoidHomomorphism", "RootGroupHomomorphism"]
gap> KnownAttributesOfObject( aut3 );
[ "Order", "Range", "Source", "SinglePieceMappingData", "ImagesOfObjects",
  "ImageElementsOfRays", "ObjectTransformationOfGroupoidHomomorphism",
  "RootGroupHomomorphism" ]
```

# **Chapter 8**

# **Development History**

# 8.1 Versions of the Package

The first version, GraphGpd 1.001, formed part of Emma Moore's thesis [Moo01] in December 2000, but was not made generally available.

Version 1.002 of GraphGpd was prepared to run under GAP 4.4 in January 2004; was submitted to the GAP council to be considered as an accepted package; but suggestions from the referee were not followed up.

In April 2006 the manual was converted to GAPDoc format. Variables Star, Costar and CoveringGroup were found to conflict with usage in other packages, and were renamed VertexStar, VertexCostar and CoveringGroupOfGroupoid respectively. Similarly, the Vertices and Arcs of an FpWeightedDigraph were changed from attributes to record components.

In the spring of 2006 the package was extensively rewritten and renamed **Gpd**. Version 1.01 was submitted as a deposited package in June 2006. Version 1.03, of October 2007, fixed some file protections, and introduced the test file gpd\_manual.tst.

Version 1.05, of November 2008, was released when the website at Bangor changed.

Since then, the package has been rewritten again, introducing magmas with objects and their mappings. Functions to implement constructions contained in [AW10] have been added, but this is ongoing work.

Versions 1.09 to 1.15 were prepared for the anticipated release of GAP 4.5 in June 2012.

Gpd became an accepted GAP package in May 2015.

In April 2017 the package was renamed again, as groupoids.

In August 2017 the implementation of groupoid homomorphisms was completely revised with the emphasis now on a mapping from a set of generating arrows to their images.

In September 2017 various functions were revised so that, at last, the operation DiscreteNormalPreXModWithObjects in XMod works again. This constructs a crossed module of groupoids with a connected range and a homogeneous, discrete source.

In recent versions there have been a number of changes of function name, such as IsDigraph becoming IsGroupoidDigraph. This is in order to avoid conflicts with the Digraphs package. The intention is that these functions can revert to the original names in due course.

In version 1.62 of October 2018 there were significant changes to the operations constructing free products with amalgamation and HNN extensions. There was a plan to move this material to a new package Rewriting, but that has not happened.

Version 1.71 of August 2022 contains a complete revision of right, left and double cosets of groupoids. (The initial declaration of LeftCoset was moved to the Utils package.)

# 8.2 What needs to be done next?

- more work on automorphism groups of groupoids;
- normal subgroupoids and quotient groupoids;
- more methods for morphisms of groupoids, particularly when the range is not connected;
- ImageElm and ImagesSource for the cases of groupoid morphisms not yet covered;
- Enumerator for IsHomsetCosetsRep;
- free groupoid on a graph;
- convert GraphOfGroupsRewritingSystem to the category IsRewritingSystem;
- in XMod, continue to work on crossed modules over groupoids.

# References

- [AW10] M. Alp and C. D. Wensley. Automorphisms and homotopies of groupoids and crossed modules. *Applied Categorical Structures*, 18:473–495, 2010. 2, 5, 32, 49, 68
- [BMPW02] R. Brown, E. J. Moore, T. Porter, and C. D. Wensley. Crossed complexes, and free crossed resolutions for amalgamated sums and hnn-extensions of groups. *Georgian Math. J.*, 9:623–644, 2002. 5
- [Bro88] R. Brown. *Topology: a geometric account of general topology, homotopy types, and the fundamental groupoid.* Ellis Horwood, Chichester, 1988. 6, 18
- [Bro06] R. Brown. *Topology and groupoids*. Booksurge LLC, S.Carolina, 2006. 6, 18
- [GH17] S. Gutsche and M. Horn. AutoDoc Generate documentation from GAP source code (Version 2017.09.15), 2017. GAP package, https://github.com/gap-packages/AutoDoc. 2
- [Hig76] P. Higgins. The fundamental groupoid of a graph of groups. *J. London Math. Soc.*, 13:145–149, 1976. 55
- [Hig05] P. Higgins. *Categories and groupoids*. Reprints in Theory and Applications of Categories, 2005. http://www.tac.mta.ca/tac/reprints/articles/7/tr7abs.html.
- [Hor17] M. Horn. GitHubPagesForGAP Template for easily using GitHub Pages within GAP packages (Version 0.2), 2017. GAP package, https://gap-system.github.io/GitHubPagesForGAP/. 2
- [LN17] F. Lübeck and M. Neunhöffer. *GAPDoc (version 1.6)*. RWTH Aachen, 2017. GAP package, http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/index.html.
- [Moo01] E. J. Moore. Graphs of Groups: Word Computations and Free Crossed Resolutions. PhD thesis, University of Wales, Bangor, 2001. http://www.maths.bangor.ac.uk/research/ftp/theses/moore.ps.gz. 5, 55, 58, 59, 68
- [Ser80] J. Serre. Trees. Springer-Verlag, Berlin, 1980. 55

# **Index**

* for groupoid elements, 24	${\tt GeneratorsOfMonoidWithObjects}, 11$
\^, 32	GeneratorsOfSemigroupWithObjects, 10
\^	GraphOfGroupoids, 62
for arrows, 32	GraphOfGroupoidsOfWord,64
for groupoids, 32	GraphOfGroupoidsWord,64
_	GraphOfGroups, 56
Arrow, 7	GraphOfGroupsOfWord, 58
Arrow	GraphOfGroupsRewritingSystem, 59
for groupoid elements, 24	GraphOfGroupsWord, 58
AutomorphismGroup, 49	Groupoid, 18
AutomorphismGroupOfGroupoid, 49	GroupoidAutomorphismByGroupAuto, 47
AutomorphismGroupoidOfGroupoid, 52	GroupoidAutomorphismByGroupAutos, 51
ConjugateGroupoid, 33	GroupoidAutomorphismByObjectPerm, 47
Cosets (left,right,double), 30	GroupoidAutomorphismByRayShifts, 47
costar, 25	GroupoidByIsomorphisms, 33
Costai, 25	GroupoidElement, 24
DigraphOfGraphOfGroupoids,62	GroupoidHomomorphism, 37
DigraphOfGraphOfGroups, 56	GroupoidHomomorphismFromHomogeneous-
DirectProductOp, 23	Discrete, 44
DiscreteSubgroupoid, 29	GroupoidHomomorphismFromSinglePiece, 37
DiscreteTrivialSubgroupoid, 28	GroupoidInnerAutomorphism, 47
DomainWithSingleObject	GroupoidsOfGraphOfGroupoids, 62
for groups, 18	GroupsOfGraphOfGroups, 56
for semigroups, 9	
double coset, 30	HeadOfArrow
DoubleCoset, 30	for groupoids, 24
DoubleCosetRepresentatives, 30	for magmas with objects, 8
DoubleCosets, 30	${\tt HeadOfGraphOfGroupsWord,58}$
	HnnExtension, 61
ElementOfArrow	HnnExtensionInfo, 61
for groupoids, 24	HomogeneousDiscreteGroupoid, $22$
for magmas with objects, 7	HomogeneousGroupoid, 22
Embedding, 23, 61	${\tt HomomorphismByUnion}$
FpWeightedDigraph, 55	for groupoids, 45
FreeProductWithAmalgamation, 59	for magmas with objects, 16
<del>-</del>	HomomorphismFromSinglePiece, 13
FreeProductWithAmalgamationInfo, 59 FullTrivialSubgroupoid, 28	${\tt HomomorphismToSinglePiece}$
	for groupoids, 43
GeneratorsOfMagmaWithObjects, 10	for magmas with objects, 13

Homset, 25	IsMappingToSinglePieceRep, 13
	IsMappingWithObjectsByFunction, 17
identity subgroupoid, 28	IsMatrixGroupoid, 20
IdentityArrow, 24	${\tt IsMultiplicative Element With Objects, 8}$
IdentityMapping, 43	IsomorphismGroupoids, 46
ImageElementsOfRays, 40	IsomorphismNewObjects
ImagesOfObjects, 40	for groupoids, 41
InclusionMappingGroupoids, 41	for magmas with objects, 13
inner automorphism, 47	IsomorphismPcGroupoid, 43
inner automorphism group, 50	IsomorphismPermGroupoid, 43
inverse arrow, 24	IsomorphismsOfGraphOfGroupoids, 62
InvolutoryArcs, 55	IsomorphismsOfGraphOfGroups, 56
IsAutomorphismWithObjects	IsomorphismStandardGroupoid, 42
for groupoid homomorphisms, 39	IsPcGroupoid, 20
IsAutomorphismWithObjects, 16	IsPermGroupoid, 20
IsBijectiveOnObjects	IsReducedGraphOfGroupoidsWord, 64
for groupoid homomorphisms, 39	IsReducedGraphOfGroupsWord, 59
IsBijectiveOnObjects, 16	IsSinglePiece, 8
${\tt IsDirectProductWithCompleteDigraph, 8}$	IsSinglePieceDomain, 8
${\tt IsDiscreteMagmaWithObjects}, 8$	IsSubgroupoid, 26
<pre>IsDomainWithObjects, 7</pre>	IsSurjective
IsEndomorphismWithObjects	for groupoid homomorphisms, 39
for groupoid homomorphisms, 39	IsSurjectiveOnObjects
IsEndomorphismWithObjects, 16	for groupoid homomorphisms, 39
IsFpGroupoid, 20	IsSurjectiveOnObjects, 16
IsFpWeightedDigraph, 55	IsWideSubgroupoid, 26
IsFreeGroupoid, 20	
IsFreeProductWithAmalgamation, 59	LeftCoset, 30
IsGraphOfFpGroupoids, 62	LeftCosetRepresentatives, 30
IsGraphOfFpGroups, 57	LeftCosetRepresentativesFromObject, 30
IsGraphOfGroupoidsWord, 64	LeftCosets, 30
IsGraphOfGroupsWord, 58	${\tt LeftTransversalsOfGraphOfGroupoids,62}$
IsGraphOfPcGroups, 57	LeftTransversalsOfGraphOfGroups, 58
IsGraphOfPermGroupoids, 62	loop, 25
IsGraphOfPermGroups, 57	
IsGroupoidByIsomorphisms, 33	MagmaWithObjects, 6
<pre>IsGroupoidWithMonoidObjects, 35</pre>	MagmaWithObjectsHomomorphism, 13
IsHnnExtension, 61	MappingToSinglePieceData
IsHomogeneousDomainWithObjects, 22	for groupoids, 37
IsHomogeneousDiscreteGroupoidRep, 22	for magmas with objects, 13
IsInjective	MappingWithObjectsByFunction, 17
for groupoid homomorphisms, 39	matrix representation, 53
IsInjectiveOnObjects	MaximalDiscreteSubgroupoid, 29
for groupoid homomorphisms, 39	MonoidWithObjects, $10$
IsInjectiveOnObjects, 16	NicoObject AutoCrounCrouncid 40
IsMagmaWithObjects, 7	NiceObjectAutoGroupGroupoid, 49 NormalFormGGRWS, 59
<pre>IsMappingToGroupWithGGRWS, 61</pre>	NOT MATE OT MAGIEWS, 37

ObjectCostar, 25	SinglePieceSemigroupWithObjects, 9
ObjectGroup, 19	SinglePieceSubgroupoidByGenerators, 29
ObjectGroupHomomorphism, 40	Size, 20
ObjectList	Source, 13
for groupoids, 19	standard groupoid, 19
for magmas with objects, 6	star, 25
ObjectList	Subgroupoid, 26
for groupoids, 20	SubgroupoidByObjects, 27
ObjectStar, 25	SubgroupoidByPieces, 27
Order, 25	SubgroupoidBySubgroup, 27
	SubgroupoidsOfGraphOfGroupoids, 62
ParentMappingGroupoids, 41	SubgroupoidWithRays, 26
PieceIsomorphisms, 22	
PieceOfObject, 11	TailOfArrow
Pieces	for groupoids, 24
for groupoids, 20	for magmas with objects, 8
for magmas with objects, 11	TailOfGraphOfGroupsWord, 58
PiecesOfMapping, 13	tree groupoid, 28
Projection, 23	trivial subgroupoid, 28
Range, 13	UnderlyingFunction, 17
RayArrowsOfGroupoid, 26	UnionOfPieces
rays, 18	for groupoids, 20
RaysOfGroupoid, 26	for magmas with objects, 11
ReducedGraphOfGroupoidsWord, 64	for magmas with objects, 11
ReducedGraphOfGroupsWord, 59	WordOfGraphOfGroupoidsWord, 64
ReducedImageElm, 61	WordOfGraphOfGroupsWord, 58
ReplaceOnePieceInUnion, 20	•
representation by matrices, 53	
RestrictedMappingGroupoids, 41	
RightActionGroupoid, 35	
RightCoset, 30	
RightCosetRepresentatives, 30	
RightCosets, 30	
RightTransversalsOfGraphOfGroupoids, 62	
RightTransversalsOfGraphOfGroups, 58	
RootGroup, 19	
RootGroupHomomorphism, 40	
RootObject	
for groupoids, 19	
for magmas with objects, 6	
· ·	
SemigroupWithObjects, 9	
SinglePieceGroupoid, 18	
SinglePieceGroupoidWithRays, 35	
SinglePieceMagmaWithObjects, 6	
SinglePieceMonoidWithObjects, 10	